
1280_Microcode/.DS_Store


1280_Microcode/asm.mac
	.NLIST	BEX,SEQ,LOC
	.LIST	TTM

	.TITLE ASM  (Generic)

; A.E.D. 2901 ASSEMBLER

REV.	= 'A  ;26-JUL-79  INITIAL CREATION

;XXX	= '4  ;08-NOV-79  ADDED OUTPUT ERROR MESSAGE.
;XXX	= '5  ;13-NOV-79  ADDED CONDITIONAL ASM AND LIST.
;XXX	= '6  ;15-NOV-79  FIXED .NLIST AND .ORG
;XXX	= '7  ;16-NOV-79  FIXED .IF FOR PSUEDO'S
;XXX	= '8  ;03-DEC-79  ADDED "BUS" TO SRC FIELD
;XXX	= '9  ;05-DEC-79  FIXED MULT. INPUT FLILES

VERS::	.WORD 10.	  ;26-Feb-85  Moved field defns out of ASM1

	.MCALL	..V2..,.REGDEF
	..V2..
	.REGDEF

	.MCALL	.PRINT,.CSIGEN,.WAIT,.SETTOP
	.MCALL	.READW,.WRITW,.CLOSE,.EXIT

	.GLOBL	SPLINE,PLINE,CPLINE,BINHX,BINDEC


	.PSECT	ASM

;*********************************
;  THIS CODE IS ABSOLUTE	;*
;	.ASECT			;*
;	. = 1000		;*
;*********************************

EMT	= 52		;ERROR STATUS WORD
�
	.MACRO	FIELD,A,B,C,D,E
	.LIST	ME
	JSR	R5,SCAN		;SCAN SOURCE FOR FIELD
	.WORD	A		;0 EOL
	.WORD	B		;1 SPACE
	.WORD	C		;2 SYMBOL
	.WORD	D		;3 SPECIAL
	.WORD	E		;4 NUMBER
	.NLIST	ME
	.ENDM

;  SET A FLAG BYTE TO "ON"
	.MACRO	FLAG,A
	MOVB	#1,A
	.ENDM

;  CALL "ERR" WITH ERROR CODE
	.MACRO	ERR,A
	MOV	#A,R0
	JSR	PC,SERR
	.ENDM

;  XOR SOURCE REG,(DESTINATION REG)+
	.MACRO	XOR	A,B
	MOV	(B),-(SP)
	COM	(SP)
	BIC	A,(B)
	COM	A
	BIC	A,(SP)
	BIS	(SP)+,(B)+
	.ENDM


;  GENERATE OP-CODE TABLE ENTRY
	.MACRO	FLD,FIELD,CODE
	.NCHR	LEN,FIELD
	.ASCII	<LEN>/FIELD/<CODE>
	.ENDM

	.LIST	LOC
�
ASM::	MOV	#1000,SP	;SET STACK
	.PRINT	#HIMSG		;HELLO
	MOV	#1000,SP	;SET STACK
	MOV	#3$,R1		;INIT ROUTINE POINTER

2$:	MOV	(R1)+,R0	;GET ADDR AND ADVANCE PTR
	MOV	R1,-(SP)	;SAVE PTR

	JSR	PC,(R0)		;GO TO THE ROUTINE

	MOV	(SP)+,R1	;RESTORE PTR
	BR	2$

3$:	.WORD	GNAME		;GET FILE NAMES AND OPT.
	.WORD	IOUT		;INIT FILE OUTPUT
	.WORD	ITBLS		;INIT ASM TABLES
	.WORD	IPASS1		;INIT FOR PASS 1
	.WORD	PASSX		;DO PASS 1
	.WORD	IPASS2		;INIT FOR PASS2
	.WORD	PASSX,EPASS2	;DO AND END PASS 2
	.WORD	PSYM		;SORT AND PRINT SYM. TBL.
	.WORD	CLEAN		;DO SOME CLEANUP
	.WORD	1$		;START ALL OVER
1$:	.EXIT

�
;  GET FILE NAMES AND SWITCHES

GNAME::	JSR	PC,IOPT		;SET OPTIONS TO DEFAULT
	.PRINT	#12$		;FILES?
	.CSIGEN	@#50,#11$	;OPEN THE FILES
	MOV	R0,@#50		;SET NEW TOP LIMIT
	MOV	(SP)+,R5	;OPTION COUNT
	BEQ	9$		;NO OPTIONS
1$:	MOV	(SP)+,R4	;GET THE OPTION
	BMI	2$		;GET THE VALUE
	CLR	-(SP)		;SET FOR NO VALUE
2$:	MOV	(SP)+,R3	;OPTION VALUE
	BIC	#177600,R4	;ISOLATE THE OPTION
	MOV	#1,R0		;INIT FLAG WITH CMD. BIT
	CMP	R4,#'L		;LIST?
	BEQ	5$		;YES
	CMP	R4,#'N		;NLIST?
	BEQ	6$		;YES
3$:	.PRINT	#10$		;BAD OPTION
	JMP	ASM		;START ALL OVER

;  LIST OPTION
5$:	MOV	#201,R0		;SET LIST OPTION

;  NLIST OPTION
6$:	MOV	#ARGL,R4	;INIT ARG. LIST PTR.
7$:	MOV	R4,R2
	TST	(R4)		;END OF LIST?
	BMI	3$		;BAD OPTION
	CMP	R3,(R4)+	;GOT IT?
	BNE	7$		;NO
	SUB	R2,R4		;GET OFFSET
	TST	-(R4)
	ASR	R4		;R4=BYTE INDEX
	MOVB	R0,LSTF(R4)	;SET OPTION FLAGS
8$:	DEC	R5		;OPTIONS DONE?
	BNE	1$		;NO, DO NEXT ONE
9$:	RTS	PC		;DONE
�
12$:	.ASCII	<15><12>'FILES? '<200>
10$:	.ASCII	<15><12>'OPTION ERROR'<200>
	.EVEN

11$:	.RAD50	'ASM'		;#3 INPUT DEFAULT
	.RAD50	'OBJ'		;#0 OUTPUT DEFAULT
	.RAD50	'LIS'		;#1 OUTPUT DEFAULT
	.WORD	0		;#2 OUTPUT

ARGL:	.WORD	0		;NO ARGS.
	.RAD50	'BIN'		;BINARY FIELD
	.RAD50	'CND'		;CONDITIONAL FIELDS
	.RAD50	'LOC'		;LOCATION FIELD
	.RAD50	'SYM'		;SYMBOL TABLE
	.WORD	-1		;END OF LIST FLAG

;  FLAG FORMAT:
;  BIT 7 = 0 = NOLIST, = 1 = LIST
;  BIT 6 = DEFAULT LIST FLAG
;  BIT 0 = 0 = STATEMENT, = 1 = COMMAND

LSTF:	.BYTE	300		;LIST
BINF:	.BYTE	300		;BINARY FIELD
CNDF:	.BYTE	0		;NO CONDITIONALS
LOCF:	.BYTE	300		;LOCATION FIELD
SYMF:	.BYTE	300		;SYMBOL TABLE
	.EVEN



;  INIT OPTIONS TO DEFAULT
IOPT::	MOV	#ARGL,R5	;INIT LIST POINTER
	MOV	#LSTF,R4	;INIT FLAG POINTER
1$:	TST	(R5)+		;DONE?
	BMI	2$		;YES
	BICB	#277,(R4)	;ISOLATE BIT 6
	MOVB	(R4),R3
	ASL	R3		;MOVE DEFAULT TO BIT 7
	BISB	R3,(R4)+	;SET DEFAULT FLAG
	BR	1$		;LOOP
2$:	RTS	PC
�
;  INIT FILE OUTPUT

IOUT::	.WAIT	#3		;ANY INPUT FILE?
	BCC	1$		;YES
	.PRINT	#4$		;NO INPUT
	JMP	ASM		;START ALL OVER

1$:	CLRB	OBJF		;SET FOR NO OBJECT
	.WAIT	#0		;ANY OBJECT FILE?
	BCS	2$		;NO
	INCB	OBJF		;SET FLAG FOR OBJECT FILE
	CLRB	GSDHD		;INIT OUTPUT FLAG

2$:	CLRB	LISF		;SET FOR NO LIST FILE
	.WAIT	#1		;ANY LISTING FILE?
	BCS	3$		;NO
	INCB	LISF		;SET FOR LISTING FILE
	MOV	#LSTBUF,R0	;LISTING BUFFER
	JSR	PC,SPLINE	;INIT OUTPUT
	MOVB	#15,HDR0	;NO FF ON 1ST PAGE

3$:	RTS	PC

4$:	.ASCII	<15><12>'NO INPUT FILE'<200>
	.EVEN



;  INIT ASSEMBLER TABLES

ITBLS::	MOV	@#50,SYMS	;START OF PERM. TABLE
	CLR	@SYMS
	.SETTOP	#-2		;TAKE ALL MEMORY
	MOV	@#50,SYML	; START OF LOCAL TABLE
	CLR	@SYML
	RTS	PC
�
;  INIT FOR PASS1

	.ENABL	LSB
IPASS1::
	CLRB	PASS		;FLAG PASS 1
IPASS::
	JSR	PC,UNBLD	;INIT FLAGS
	CLRB	ENDF		;PROGRAM END FLAG
	CLRB	COND		;RESET CONDITIONAL FLAG
	CLRB	CONDS		;ALSO SAVED COND. FLAG
	CLRB	NLST		;RESET NOLIST FLAG
	MOV	SYML,SYMLS	;RESET LOCAL SEARCH
	MOVB	#3,FILE		;SET STARTING FILE NO.
	CLR	RBLK		;INIT READ BLOCK NO.
	CLRB	EOFCNT		;NO EOF'S YET
	MOV	#77777,RPTR	;SET TO READ 1ST SECTOR
	MOV	LOC,MLOC	;SAVE MAX. LOC.
	DEC	MLOC		;ADJUST TO LAST USED
	CLR	LOC		;INIT THE PC
	JSR	R5,GETL		;GET A LINE
LINEP::
	.WORD	LINE,3$		;LINE BUFFER, INPUT ERROR
	FLAG	PSUDOF		;START WITH PSUEDO OP LISTING
	JSR	PC,PROC		;PROCESS THE LINE
	RTS	PC



;  ASSEMBLER PASS 1

PASSX::
	TSTB	ENDF		;EOF OR END STATEMENT?
	BNE	2$		;YES, STOP
1$:	JSR	R5,GETL		;GET A LINE
	.WORD	LINE,3$		;LINE BUFFER, INPUT ERROR
	JSR	PC,PROC		;PROCESS THE LINE
	JSR	PC,LIST		;LIST PREVIOUS LINE
	TSTB	ENDF		;EOF OR END STATEMENT?
	BEQ	1$		;NO, DO NEXT
2$:	RTS	PC		;DONE
3$:	.PRINT	#4$		;ERROR
	.EXIT
4$:	.ASCII	<15><12>'INPUT ERROR'<200>
	.EVEN
	.DSABL	LSB



;  INIT FOR PASS 2
IPASS2::
	INCB	PASS		;SET FOR PASS 2
	CLR	ERCNT		;INIT ERROR COUNT
	CLR	PAGE
	JSR	PC,IPASS	;GET SOURCE READY
	MOVB	#177,LCNTR	;SET FOR PAGE EJECT
	MOVB	#-1,LCNTR	;SET FOR NEW PAGE
	JSR	PC,CLRLST	;CLEAR LISTING LINE
	RTS	PC
�
;  CLOSE THE OBJECT OUTPUT FILE
EPASS2::
	MOV	#LINEP,R5
	JSR	PC,CPYLST	;RDY FOR LAST LINE
	JSR	PC,BLST		;BUILD THE LISTING
	JSR	PC,LIST		;LIST LAST LINE
	TSTB	OBJF		;OUTPUT?
	BEQ	5$		;NO
	MOV	BBPTR,R4
	MOV	BINPTR,R1	;CURRENT BIN BLOCK ADRS.
	SUB	#BINBLK,R1	;CALCULATE BYTE COUNT.
	MOVB	CKSUM,R3
	ADD	R1,R3		;ADD TO CHECKSUM
	MOVB	R3,CKSUM
	MOVB	R1,@#BINBLK+2	;SET BYTE COUNT IN BUF.
	MOV	#BINBLK,R0
1$:	JSR	PC,PUTB		;TEST BUF FULL.
	MOVB	(R0)+,(R4)+
	DEC	R1		;LAST BYTE?
	BGT	1$		;NO
	NEGB	CKSUM		;SET CKSUM NEG.
	MOV	#CKSUM,R0	;PUT ENDMOD BLOCK IN BUF.
2$:	JSR	PC,PUTB		;TEST BUF FULL.
	MOVB	(R0)+,(R4)+
	CMP	R0,#EENDBK	;END OF BLOCK?
	BLO	2$		;NO
3$:	CMP	R4,#EBINBF	;BUF FULL?
	BHIS	4$		;YES
	MOVB	#0,(R4)+	;FILL REMANING WITH 0.
	BR	3$
4$:	MOV	#BPAR,R0
	.WRITW		;WRITE LAST BLOCK IN BUF.
	.CLOSE	#0	;CLOSE THE BINARY OUTPUT FILE.
5$:	.CLOSE	#3	;CLOSE INPUT
	RTS	PC

;  PRINT SYMBOL TABLE
PSYM::
	RTS	PC

;  FINAL CLEANUP
CLEAN::
	MOV	ERCNT,R3	;GET ERROR COUNT
	MOV	#ERLN2,R4	;ASCII DESTINATION
	MOV	#4,R5		;NO. CHAR.
	JSR	PC,BINDEC	;CONVERT TO DECIMAL
	JSR	PC,LINPG	;CHECK FOR NEW PAGE
	MOV	#ERLN,R0
	JSR	PC,PRNT		;PRINT ERRORS
	MOV	#ERLN,R0
	.PRINT			;PRINT ERROR COUNT
	TSTB	LISF		;LIST FILE?
	BEQ	1$		;NO
	JSR	PC,CPLINE	;CLOSE LISTING FILE
	.CLOSE	#1		;CLOSE LISTING FILE
1$:	RTS	PC
ERLN:	.ASCII	<15><12>'ERRORS = '
ERLN2:	.ASCII	'NNNN.'<200>
	.EVEN
�
;  GET A SOURCE LINE
;	JSR	R5,GETL
;	.WORD	LINE BUFFER
;	.WORD	ERROR

GETL::
	JSR	PC,CPYLST	;MOVE OLD LINE TO LISTER
	MOV	RPTR,R1		;GET SOURCE POINTER
	MOV	#RBUFF+512.,R2	;END OF BUFF POINTER
	MOV	(R5)+,R0	;LINE POINTER
	MOV	R0,R3
	ADD	#79.,R3		;END OF LINE POINTER
	MOVB	#200,(R0)	;MARK THE END OF LINE
1$:	CMP	R0,R3		;END OF OUPUT LINE?
	BNE	2$		;NO
	MOVB	#200,-(R0)	;YES, BACK UP ONE
2$:	CMP	R1,R2		;END OF INPUT BUFFER?
	BLO	23$		;NO
	JSR	R5,6$		;YES, GET ANOTHER SECTOR
	.WORD	5$		;INPUT ERROR
23$:	MOVB	(R1)+,R4	;GET A BYTE
	BEQ	2$		;SKIP ZEROS
	CMPB	#12,R4		;LF?
	BEQ	2$		;SKIP LF'S
	CMPB	#14,R4		;FF?
	BNE	3$		;NO, LOOP
	TSTB	LCNTR		;RDY TO EJECT?
	BMI	2$		;YES, SKIP IT
	CLRB	LCNTR		;NO, SET TO EJECT
	BR	2$		;THEN SKIP IT
3$:	MOVB	R4,(R0)+	;MOVE BYTE TO LINE BUFF
	CMPB	#15,R4		;CR?
	BNE	1$		;NO, LOOP
	MOVB	#200,-(R0)	;MARK END OF LINE
	MOV	R1,RPTR
4$:	TST	(R5)+		;SET FOR OK EXIT
	RTS	R5
5$:	MOV	(R5),R5		;SET FOR ERROR EXIT
	RTS	R5

;  READ A SECTOR
6$:	MOV	R0,-(SP)
7$:	MOVB	FILE,RPAR	;SET FILE #
	MOV	#RPAR,R0	;READ PARAMETERS
	.READW			;READ A SECTOR
	BCS	9$		;ERROR
	INC	RBLK		;NEXT BLOCK
	MOV	#RBUFF,R1	;RESET BUFFER POINTER
8$:	MOV	(SP)+,R0
	BR	4$		;OK EXIT
9$:	MOV	@#EMT,R0	;ERROR STATUS (0=EOF)
	BEQ	15$		;EOF
	CMP	R0,#2		;FILE NOT OPEN?
	BEQ	12$		;TRUE
10$:	TST	(SP)+		;POP
	BR	5$		;NOT EOF = ERROR EXIT
�
;  OPEN THE NEXT FILE
15$:	CMPB	FILE,#8.	;LAST FILE?
	BHIS	11$		;YES
	INCB	FILE		;SET FOR NEXT FILE
	.WAIT	FILE		;FILE THERE?
	BCS	11$		;NO
	CLR	RBLK		;INIT THE FILE
	BR	7$		;GO BACK AND READ A SECT.
11$:	DECB	FILE		;MOVE BACK TO LAST
	TSTB	EOFCNT		;FIRST TIME FOR EOF?
	BNE	12$		;NO
	INCB	EOFCNT		;SET TO SECOND TIME
	MOVB	#15,-(R1)	;SET CR FOR END OF LINE
	MOV	(SP)+,R0
	BR	4$		;TAKE OK EXIT
12$:	CMPB	EOFCNT,#1	;SECOND TIME FOR EOF?
	BNE	10$		;NO = ERROR
;  SHOW AN "END" STATEMENT
	MOV	#16$-14$,-(SP)	;CHARACTER COUNTER
	MOV	#16$,R0		;"END" POINTER
13$:	MOVB	-(R0),-(R1)	;MOVE THE "END"
	DEC	(SP)		;DONE?
	BNE	13$		;NO
	TST	(SP)+		;POP THE COUNTER
	BR	8$		;TAKE OK EXIT
14$:	.ASCII	' .END 0'<15>
16$:
	.EVEN


;  COPY A LINE TO LISTER LINE
CPYLST::
	MOV	#ERRS,R0	;ERROR STORAGE
	MOVB	.BCN,R1		;WORD BIT LENGTH
	ASR	R1		;/2
	ASR	R1		;/4 = CHAR. CNT.
	ADD	#CLINE2,R1	;LISTING OUTPUT LINE
	MOVB	#40,(R1)+	;SET A SPACE
	MOV	R1,STLIN	;SAVE START OF LINE
	MOV	#40.,R2		;80. BYTES
	MOV	(R0),CLINE+2	;COPY ERROR CODES
	MOV	#20040,(R0)+	;CLEAR ERRORS
20$:	MOVB	(R0)+,(R1)+	;MAKE LISTING LINE
	BMI	21$		;DONE
	MOVB	(R0)+,(R1)+
	BMI	21$		;DONE.
	DEC	R2		;DONE?
	BNE	20$		;NO
21$:	RTS	PC
�
;  BEWARE MAKING CHANGES TO "PROC" THROUGH "FIGR".
;  IT IS LIKE PLAYING "PICK-UP-THE-STICKS" IN AN EARTHQUAKE.
;  ALSO KNOWN AS "A CAN OF WORMS".  NEEDS TO BE REWRITTEN
;  IN THE WORST WAY, IF THE TIME CAN JUST BE FOUND.

;  PROCESS A STATEMENT LINE
PROC::
;
;INIT FOR PROCESSING
;
	MOV	#LINE,SPTR	;INIT SCAN POINTER
	CLRB	PCNTF		;CONTINUATION FLAG
	CLRB	PLBLF		;LABEL FLAG
	CLRB	PERRF		;ERROR FLAG
	CLRB	POBJF		;OBJECT FLAG
	CLRB	PEQUF		;EQUATE FIELD FLAG
	CLRB	POUTF		;OBJECT OUT FLAG
	CLRB	PNOP		;NOP FLAG
	CLRB	PEQUS		;= FLAG
	CLRB	PDEST		;DESTIATION FLAG
	CLRB	PLFLG		;LINE FLAG
	CLRB	PJADDR
	CLRB	PCOND
	CLRB	PFNOP		;FUNCTION NOP
	CLRB	PCOMA		;COMMA FLAG
	CLRB	PRVAL		;R VALUE FLAG
	CLRB	PSRC		;SOURCE FLAG
	CLRB	PMRF		;-R FLAG
	CLRB	PDTA1		;DATA FLAG
	CLRB	PDTA3		;DATA FLAG
	CLRB	PSVAL		;S VALUE FLAG
	CLRB	PSPEC		;SPECIAL OP CODE FLAG
	CLRB	PRV
	CLRB	PSV
	CLRB	PCRY		;CARRY FLAG
	CLRB	PCRRY		;CARRY FLAG
	CLRB	PJCND		;JUMP CONDITION FLAG
	CLRB	PAADDR
	CLRB	PBADDR

	JSR	R5,PRLBL	;PROCESS LABEL FIELD
	.WORD	3$		;COMMENT

	JSR	R5,PROP		;PROCESS OPERATION
	.WORD	3$		;COMMENT
	.WORD	4$		;JUMP CODE
	.WORD	2$		;EQUATE FIELD
	.WORD	1$		;= FLAG IS ON

	JSR	R5,PRMOD	;DO OPERATION MODIFIER
	.WORD	3$		;COMMENT EXIT

1$:	JSR	PC,PRFUN	;DO OPERATION FUNCTION

2$:	JSR	PC,PREQU	;PROCESS EQUATE FIELD

3$:	RTS	PC

4$:	TSTB	PCNTF		;CONTINUATION?
	BNE	6$		;YES
	MOVB	#1,DEST		;DEST = NOP
	FLAG	PDEST
	MOVB	#4,FUNC		;FUNC = NOP
	FLAG	PFUNC
	MOVB	#2,SORC		;SORC = NOP
	FLAG	PSRC
6$:	JSR	R5,PRADD	;PROCESS JUMP ADDRESS
	.WORD	3$		;COMMENT.
	BR	2$		;PROCESS EQUATE FIELD
�
;  PROCESS THE LABEL FIELD
;	JSR	R5,PRLBL	;PROCESS THE LABEL
;	.WORD	COMMENT EXIT

PRLBL::
	FIELD	1$,4$,3$,6$,1$	;CHECK FOR LABEL FIELD
1$:	JSR	PC,DOBJ		;ERR, WRITE LAST WORD
	MOVB	#-1,PSUDOF	;SET COMMENT FLAG
	MOV	(R5),R5		;SET EXIT ADDRESS
	RTS	R5		;EXIT

3$:	JSR	PC,DOBJ		;OUTPUT LAST STATEMENT
	JSR	PC,SLBL		;ENTER LABLE IN SYM TBL
	BR	5$		;NORM. EXIT

4$:	JSR	PC,DOBJ		;OUTPUT LAST STATEMENT

5$:	TST	(R5)+		;ADJUST FOR NORMAL EXIT
	RTS	R5

6$:	CMP	R1,#'*		;COMMENT?
	BEQ	1$		;YES, TAKE COMMENT EXIT
	CMP	R1,#';		;COMMENT?
	BEQ	1$		;YES, TAKE COMMENT EXIT
	CMP	R1,#16		;^N?
	BEQ	1$		;YES
	CMP	R1,#'+		;CONTINUATION?
	BEQ	7$		;YES
	ERR	'L		;NO = LABEL ERROR
	BR	1$		;ERROR EXIT

7$:	FLAG	PCNTF		;SET CONTINUATION FLAG
	BR	5$		;IN-LINE EXIT
�
;  PROCESS OPCODE FIELD

;	JSR	R5,PROP
;	.WORD	COMMENT EXIT
;	.WORD	JUMP CODE ADDRESS
;	.WORD	EQUATE FIELD ADDRESS
;	.WORD	"=" SIGN ADDRESS

PROP::
	CLRB	PSUDOF		;CLEAR COMMENT FLAG
	FIELD	61$,PROP,2$,10$,15$
61$:	MOVB	#-1,PSUDOF	;SET COMMENT FLAG
1$:	MOV	(R5),R5		;EXIT
	RTS	R5

2$:	JSR	R5,SRC		;SEARCH FOR JUMP CODE
	.WORD	JSRC,4$		;JMP TBL, NOT FOUND ADDR
	TSTB	COND		;CODE ON?
	BNE	1$		;NO
;  DO THE JUMP CODE
	BIT	#20,R1		;ADDRESS?
	BEQ	60$		;NO
	FLAG	PJADDR		;YES, FLAG IT
60$:	BIC	#177760,R1	;ISOLATE JUMP CODE
	MOVB	R1,SCONT	;SET SEQU. CONT.
	FLAG	PSCONT		;FLAG IT
	TST	(R5)+		;READY FOR JUMP EXIT
	BR	1$

4$:	JSR	R5,SRC		;SEARCH FOR PSUEDO OP
	.WORD	PSRCH,5$	;PSU OP TBL, NOT FND ADDR

	JSR	PC,PSUDO	;DO THE PSUEDO OP
	BR	1$		;TAKE COMMENT EXIT

5$:	TSTB	COND		;CODE ON?
	BNE	1$		;NO
	JSR	R5,SRC		;SEARCH FOR DESTINATION
	.WORD	OPSRC,8$		;DEST. TBL, NOT FND ADDR

	MOVB	R1,DEST		;SET DESTINATION
	BEQ	63$		;NOP?
	FLAG	PNOP		;FLAG IT
63$:	FLAG	PDEST		;FLAG THE DESTINATION
	ADD	#8.,R5		;SET FOR IN-LINE EXIT
	RTS	R5

8$:	JSR	R5,LKUP		;FIND SYMBOL
	.WORD	17$		;ERROR OR NOT FOUND ADDRESS
	JSR	PC,9$		;SET THE DEST, FLG AND EXIT
	RTS	R5
�
; SET DESTINATION VALUE, FLAG IT, AND SET EXIT
9$:	MOVB	R1,BADDR	;SET DESTINATION ADDRESS
	FLAG	PBADDR		;FLAG THE B ADDRESS
	MOVB	#3,DEST		;SET DESTINATION
	FLAG	PDEST		;FLAG IT
	ADD	#8.,R5		;SET FOR IN-LINE EXIT
	RTS	PC

10$:	CMP	R1,#':		;":"?
	BEQ	PROP		;YES, SKIP IT
	TSTB	COND		;CODE ON?
	BNE	1$		;NO
	CMP	R1,#'(		;EXPRESSION?
	BNE	11$		;NO
	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
	.WORD	1$		;ERROR ADDRESS

	JSR	PC,9$		;SET DEST., FLAG IT, AND EXIT
	RTS	R5

11$:	CMP	R1,#'#		;EQUATE FIELD?
	BNE	13$		;NO
	FLAG	PEQUF		;SET EQUATE FLAG
	ADD	#4,R5		;SET EQUATE EXIT
	BR	1$

13$:	CMP	R1,#'=		;"=" SIGN?
	BNE	14$		;NO
	FLAG	PEQUS		;SET "=" SIGN FLAG
	MOVB	#1,DEST		;SET DEST. = NOP
	FLAG	PNOP		;FLAG IT
	ADD	#6,R5		;SET FOR "=" EXIT
	JMP	1$

14$:	CMP	R1,#'*		;COMMENT?
	BEQ	51$		;YES
	CMP	R1,#';		;COMMENT?
	BNE	17$		;NO
51$:	JMP	61$		;YES

17$:	ERR	'D		;DESTINATION ERROR
	BR	1$		;TAKE COMMENT EXIT

15$:	TSTB	COND		;CODE ON?
	BNE	51$		;NO
	JSR	R5,EVNUM	;EVALUATE NUMBER
	.WORD	1$

	JSR	PC,9$		;SET DEST, FLAG, EXIT
	RTS	R5		;TAKE EXIT
�
;  PROCESS OPERATION MODIFIER FIELD

;	JSR	R5,PRMOD
;	.WORD	COMMENT EXIT ADDRESS

PRMOD::
	FIELD	2$,2$,4$,6$,3$
1$:	MOV	(R5),R5		;ERROR, USE COMMENT EXIT
	RTS	R5

2$:	FLAG	PFNOP		;FLAG FUNCTION NOP
	TST	(R5)+		;SET FOR IN-LINE EXIT
	RTS	R5

3$:	ERR	'A		;DEST. MODIFIER ERROR
	BR	1$

4$:	TSTB	PCOMA		;COMMA FLAG ON?
	BEQ	3$		;NO
	JSR	R5,SRC		;SEARCH FOR OPERATION

	.WORD	MSRC,3$		;MOD. TBL, NOT FND ADDR
	CMP	R1,#2		;"OUT"?
	BNE	5$		;NO

	CLRB	PCOMA		;YES, CLEAR COMMA FLAG
	FLAG	POUTF		;SET "OUT" FLAG
	BR	PRMOD		;GET NEXT FIELD

5$:	CMP	R1,#4		;MODIFIER?
	BLO	3$		;NO
	CLRB	PCOMA		;YES, CLEAR COMMA FLAG
	BISB	#1,SHIFT	;SET SHIFT CONTROL
	BIC	#177774,R1	;CLEAR SHIFT BIT
	MOVB	R1,DEST		;SET DESTINATION
	FLAG	PDEST		;FLAG IT
	FLAG	PEQUS		;SET "=" SIGN FLAG
	BR	PRMOD		;GO LOOK FOR "="
�
;  SPECIAL CHARACTER FOUND
6$:	TSTB	PCOMA		;COMMA FLAG SET?
	BNE	3$		;YES = ERROR
	CMP	R1,#'=		;"=" SIGN?
	BNE	7$		;NO
	FLAG	PEQUS		;YES, FLAG IT
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5

7$:	TSTB	PEQUS		;"=" FLAG?
	BNE	3$		;YES = ERROR
	TSTB	POUTF		;"OUT" FLAG?
	BEQ	8$		;NO
	CMP	R1,#'(		;START OF EXPRESSION?
	BNE	3$		;NO = ERROR
	TSTB	PBADDR		;DESTINATION ADDR. SET?
	BEQ	3$		;NO = ERROR
	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
	.WORD	1$		;ERROR

	MOVB	R1,AADDR	;SET "A" ADDRESS
	FLAG	PAADDR		;FLAG IT
	MOVB	#2,DEST		;SET "OUT(A)" DEST. CODE
	FLAG	PEQUS		;FLAG TO REQUIRE "="
	BR	PRMOD		;GO FIND "="

8$:	CMP	R1,#',		;COMMA?
	BNE	3$		;NO
	FLAG	PCOMA		;YES, FLAG IT
	BR	PRMOD		;GO GET MODIFIER
�
;  PROCESS "R FUNCTION S"
;	JSR	PC,PRFUN

PRFUN::
	JSR	R5,PRF1		;PROCESS FIRST TERM
	.WORD	1$,2$,3$	;EXIT FOR F3, CRY, ERROR

	JSR	R5,PRF2		;PROCESS FUNCTION CODE
	.WORD	2$,3$		;EXIT FOR CRY OR ERROR

1$:	JSR	R5,PRF3		;PROCESS SECOND TERM
	.WORD	3$		;END OF FUNCTION

2$:	JSR	PC,PRCRY	;PROCESS CARRY FIELD

3$:	JSR	PC,FIGR		;FIGURE OUT "R AND S"
	RTS	PC




;	JSR	R5,PRF1		;PROCESS 1ST TERM
;	.WORD	3RD TERM EXIT ADDRESS
;	.WORD	CARRY EXIT ADDRESS
;	.WORD	ERROR EXIT ADDRESS
;  IN-LINE EXIT FOR 2ND TERM

PRF1::
	CLRB	PNEGF		;INIT TO NOT NEGATIVE
	TSTB	PFNOP		;FUNC = NOP?
	BEQ	63$		;NO
	MOV	#7,R1		;YES, GET NOP CODE
	BR	11$		;GO SET NOP

63$:	FIELD	1$,1$,6$,2$,5$

1$:	ERR	'T		;TERM ERROR
	JMP	17$		;TAKE ERROR EXIT

2$:	CMP	R1,#'+		;PLUS?
	BEQ	63$		;YES, SKIP IT

	CMP	R1,#'-		;MINUS?
	BNE	3$		;NO

	TSTB	PDTA1		;YES, IMMEDIATE DATA?
	BNE	62$		;YES

	FLAG	PMRF		;FLAG MINUS
	BR	63$		;GET THE TERM

62$:	FLAG	PNEGF		;FLAG IT AS NEGATIVE
	BR	63$		;NOW GET THE TERM

3$:	CMP	R1,#'#		;IMMEDIATE DATA?
	BNE	4$		;NO

	TSTB	PDTA1		;ALLREADY FLAGED?
	BNE	1$		;YES = ERROR

	FLAG	PDTA1		;NO, FLAG IT
	BR	63$		;GET THE DATA
�
4$:	JSR	R5,CKPC		;IS IT THE PC?
	.WORD	21$		;YES RETURN

	CMP	R1,#'(		;EXPRESSION?
	BNE	1$		;NO = ERROR

	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
	.WORD	17$		;ERROR
	BR	31$		;GO SET "R" ETC.

5$:	JSR	R5,EVNUM	;EVALUATE THE NUMBER
	.WORD	17$		;ERROR

31$:	JMP	21$		;GO SET "R" ETC.

6$:	JSR	R5,SRC		;LOOK FOR 1ST TERM
	.WORD	F1SRC,20$	;TABLE, NOT FOUND ADDR
	TSTB	PDTA1		;IMMEDIATE DATA?

	BNE	1$		;YES = ERROR
	CMP	R1,#10		;SPECIAL OP-CODE?
	BLO	11$		;NO

	TSTB	PMRF		;YES, MINUS FLAG?
	BNE	1$		;YES = ERROR

	CMP	R1,#10		;"INC"
	BNE	7$		;NO

	MOVB	.SEC,CCONT	;SET TO SET CARRY
	BR	8$

7$:	CMP	R1,#11		;"DEC"
	BNE	9$		;NO

	MOVB	.CLC,CCONT	;SET TO CLEAR CARRY
8$:	FLAG	PCRY		;FLAG CARRY
9$:	FLAG	PSPEC		;FLAG SPECIAL OP-CODE
	BIC	#177770,R1
	MOVB	R1,FUNC		;SET FUNCTION
	FLAG	PFUNC		;FLAG IT
	MOVB	#5,RV		;SET "R" = Z
	FLAG	PRV		;FLAG IT
10$:	MOV	(R5),R5		;3RD TERM EXIT
	RTS	R5

11$:	MOVB	#5,RV		;SET "R" = Z
	FLAG	PRV		;FLAG IT
	MOVB	#3,FUNC		;SET FUNC = "OR"
	FLAG	PFUNC		;FLAG FUNCTION
	CMP	R1,#2		;"Q"?
	BNE	14$		;NO

	MOVB	#2,SV		;SET "S" = Q
	FLAG	PSV		;FLAG IT
	CLRB	PRV		;TURN "R" OFF
	TSTB	PMRF		;MINUS?
	BNE	13$		;YES

	BR	15$

13$:	MOVB	#2,FUNC		;SET FUNC= "SUBS"
	MOVB	.SEC,CCONT	;SET TO SET CARRY
	FLAG	PCRY		;FLAG IT
	BR	15$
�
14$:	CMP	R1,#6		;"D"?
	BNE	30$		;NO

	FLAG	PDTA1		;YES, FLAG IT
	MOVB	RV,SV		;SET "S" = Z
	MOVB	PRV,PSV		;FLAG IT
	MOVB	R1,RV		;SET "R" = D
	BR	15$		;IN-LINE EXIT

30$:	MOVB	#2,SV		;SET "S"=Q
	FLAG	PSV		;FLAG IT
	MOVB	#4,FUNC		;SET FUNC ="AND"
	CMP	R1,#7		;NOP?
	BEQ	17$		;YES

15$:	ADD	#6,R5		;IN-LINE EXIT
	RTS	R5

17$:	CMP	(R5)+,(R5)+	;RDY FOR "ALL DONE" EXIT
	BR	10$

20$:	JSR	R5,LKUP		;LOOK FOR THE SYMBOL
	.WORD	23$		;NOT FOUND

21$:	FLAG	PRV		;FLAG "R"
	TSTB	PDTA1		;IMMEDIATE DATA?
	BNE	22$		;YES

	MOVB	R1,RVAL		;NO, SET "R" ADDRESS
	FLAG	PRVAL		;FLAG IT
	MOVB	#4,RV		;SET "R" = A
	TSTB	PBADDR		;B ADDR SET?
	BEQ	15$		;NO

	CMPB	R1,BADDR	;YES, "R" SAME AS B?
	BNE	15$		;NO

	CLRB	RV		;YES, SWITCH "R" TO B
	BR	15$		;TAKE IN-LINE EXIT

22$:	MOVB	#6,RV		;SET "R" = D
	TSTB	PNEGF		;TERM NEGATIVE?
	BEQ	24$		;NO

	NEG	R1		;YES
24$:	MOV	R1,IDTA		;SET IMMEDIATE DATA
	MOVB	.IMC,IMC	;SET IMMEDIATE CONTROL
	BR	15$		;TAKE IN-LINE EXIT

23$:	ERR	'U		;UNDEFINED
	BR	17$
�
;	JSR	R5,PRF2		;PROCESS 2ND TERM
;	.WORD	CARRY EXIT ADDRESS
;	.WORD	ERROR EXIT ADDRESS
;  IN-LINE EXIT FOR 3RD TERM

PRF2::
	FIELD	6$,5$,20$,3$,1$
1$:	ERR	'F		;FUNCTION ERROR
2$:	TST	(R5)+		;ERROR EXIT
	MOV	(R5),R5
	RTS	R5

3$:	CMP	R1,#'+		;PLUS?
	BEQ	14$		;YES

	CMP	R1,#'-		;MINUS?
	BEQ	13$		;YES

	CMP	R1,#';		;COMMENT?
	BEQ	7$		;YES

	CMP	R1,#'*		;COMMENT?
	BEQ	7$		;YES

	CMP	R1,#',		;COMMA?
	BEQ	4$		;YES

	CMP	R1,#'#		;CONTROL?
	BNE	1$		;NO = ERROR

4$:	FLAG	PEQUF		;YES FLAG IT
	BR	6$		;QUIT

5$:	SUB	R1,R2		;FIELD LENGTH
	DEC	R2		;JUST ONE SPACE OR TAB?
	BNE	6$		;NO = END OF FUNCTION

	CMPB	(R1),#40	;SPACE?
	BEQ	PRF2		;YEA, GET THE FUNCTION

6$:	TSTB	PFUNC		;FUNC ALLREADY SET?
	BNE	2$		;YES, QUIT

7$:	FLAG	PFUNC		;FLAG FUNCTION
	CMPB	RV,#6		;"R" = D?
	BNE	12$		;NO

	MOVB	#5,SV		;SET "S" = Z
	FLAG	PSV		;FLAG IT
	TSTB	PMRF		;MINUS?
	BEQ	10$		;NO

	MOVB	#1,FUNC		;SET FUNC = "SUBR"
8$:	MOVB	.SEC,CCONT	;SET TO SET CARRY
	BR	2$		;TAKE COMMENT EXIT

10$:	MOVB	#3,FUNC		;SET FUNC = "OR"
11$:	BR	2$		;TAKE COMMENT EXIT
�
12$:	MOVB	RV,SV		;MOVE "R" TO "S"
	FLAG	PSV		;FLAG "S"
	MOVB	RVAL,SVAL
	CLRB	PRVAL
	FLAG	PSVAL
	MOVB	#5,RV		;SET "R" = Z
	TSTB	PMRF		;MINUS?
	BEQ	10$		;NO, SET "OR"

	MOVB	#2,FUNC		;SET FUNC = "SUBS"
	BR	8$		;GO SET CARRY

13$:	TSTB	PMRF		;MINUS?
	BEQ	16$		;NO

	FLAG	PCRRY		;FLAG CARRY FOR "-"
	BR	7$		;PROCESS AS AN "EOL"

14$:	FLAG	PFUNC		;FLAG THE FUNCTION
	TSTB	PMRF		;MINUS?
	BNE	15$		;YES

	CLRB	FUNC		;SET FUNC = "ADD"
	BR	17$

15$:	CLRB	PMRF		;CLEAR THE MINUS
	MOVB	#1,FUNC		;SET FUNC = "SUBR"
	BR	17$

16$:	MOVB	#2,FUNC		;SET FUNC = "SUBS"
17$:	FLAG	PFUNC
	CMPB	FUNC,#2		;"+" OR "-"?
	BHI	18$		;NO = NO CARRY CONTROL

	MOVB	.CLC,CCONT	;SET TO CLEAR CARRY
	TSTB	FUNC		;FUNC = "ADD"?
	BEQ	19$		;YES, LEAVE CARRY OFF

	MOVB	.SEC,CCONT	;NO, SET CARRY ON
	TSTB	PRV		;IS THERE A "R"?
	BNE	19$		;YES, NO FUNC SWAP.

	TSTB	PSV		;"S" SET?
	BEQ	19$		;NO

	CMPB	SV,#2		;"S"=Q?
	BNE	19$		;NO

	MOVB	#1,FUNC		;SET TO "SUBR"
18$:	TSTB	PSV		;"S" SET?
	BEQ	19$		;NO

	CMPB	SV,#2		;"S"=Q?
	BNE	19$		;NO

	CMPB	FUNC,#5		;"NOTAND"
	BNE	19$		;NO

	ERR	'T		;TERM ERROR
	BR	11$		;ERROR EXIT

19$:	CMP	(R5)+,(R5)+	;MOVE TO IN-LINE EXIT
	RTS	R5

20$:	JSR	R5,SRC		;LOOK FOR OP CODE
	.WORD	F2SRC,1$	;TABLE, NOT FOUND

	MOVB	R1,FUNC		;SET FUNC
	BR	17$
�
;	JSR	R5,PRF3		;PROCESS 3RD TERM
;	.WORD	ERROR EXIT ADDRESS
;  IN-LINE EXIT FOR CARRY

PRF3::
	CLRB	PNEGF		;INIT TO NOT NEGATIVE
1$:	FIELD	2$,8$,9$,4$,7$
2$:	ERR	'T		;TERM ERROR
3$:	MOV	(R5),R5		;ERROR EXIT
	RTS	R5

4$:	CMP	R1,#'#		;IMMEDIATE DATA?
	BNE	5$		;NO

	TSTB	PDTA3		;IMMED. ALLREADY ON?
	BNE	2$		;YES = ERROR

	FLAG	PDTA3		;NO, SET IT
	BR	1$		;GO GET THE DATA

5$:	JSR	R5,CKPC		;IS IT THE PC?
	.WORD	13$		;YES RETURN.
	CMP	R1,#'(		;EXPRESSION?
	BNE	6$		;NO

	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
	.WORD	3$		;ERROR
	BR	13$

6$:	TSTB	PDTA3		;IMMEDIATE?
	BEQ	2$		;NO

	CMP	R1,#'-		;TO BE NEGATIVE?
	BNE	2$		;NO = ERROR

	FLAG	PNEGF		;YES, FLAG IT
	BR	1$		;GO GET THE TERM

7$:	JSR	R5,EVNUM	;EVALUATE NUMBER
	.WORD	3$		;ERROR
	BR	13$

8$:	SUB	R1,R2		;FIELD LENGTH
	DEC	R2		;JUST ONE SPACE?
	BEQ	1$		;YEA, GET THE TERM

	BR	2$		;NO = ERROR
�
9$:	JSR	R5,SRC		;LOOK FOR THE 3RD TERM
	.WORD	F3SRC,12$	;TABLE, NOT FND ADDR

	TSTB	PDTA3		;IMMEDIATE?
	BNE	2$		;YES = ERROR

	CMP	R1,#6		;"D"?
	BEQ	16$		;YES

	MOVB	R1,SV		;NO, SET "S"
	FLAG	PSV		;FLAG IT
	TSTB	PSPEC		;SECIAL OP CODE?
	BNE	11$		;YES

	CMP	R1,#2		;Q?
	BNE	10$		;NO

	TSTB	PRV		;"R" VALUE SET?
	BEQ	10$		;NO

	TSTB	RV		;"R" SET FOR B?
	BNE	10$		;NO

	MOVB	#4,RV		;YES, SWITCH TO A
10$:	TST	(R5)+		;CARRY EXIT
	RTS	R5

11$:	CMP	R1,#2		;"Q"?
62$:	BNE	2$		;NO = ERROR
	BR	10$		;YES = CARRY EXIT

12$:	JSR	R5,LKUP		;LOOK FOR THE SYMBOL
	.WORD	2$		;NOT FOUND

13$:	TSTB	PDTA3		;IMMEDIATE DATA?
	BNE	14$		;YES

	TSTB	PSV		;"S" TAKEN?
	BEQ	63$		;NO, GO USE IT

	TSTB	PRV		;"R" TAKEN?
	BNE	63$		;YES, GO AHEAD AND USE "S"

	MOVB	#4,RV		;SET "R" = A
	FLAG	PRV		;FLAG IT
	MOVB	R1,RVAL		;SET "R" ADDRESS
	FLAG	PRVAL		;FLAG IT
	BR	10$		;TAKE CARRY EXIT

63$:	MOVB	#4,SV		;SET "S" = A
	FLAG	PSV		;FLAG IT
	MOVB	R1,SVAL		;SET "S" ADDRESS
	FLAG	PSVAL		;FLAG IT
	BR	10$		;CARRY EXIT
�
14$:	TSTB	PNEGF		;TERM NEGATIVE?
	BEQ	15$		;NO

	NEG	R1		;YES
15$:	MOV	R1,IDTA		;SET IMMEDIATE DATA
	MOVB	.IMC,IMC	;SET DATA CONTROL CODE
16$:	TSTB	PDTA1		;1ST TERM IMMEDIATE?
	BNE	62$		;YES = ERROR

	TSTB	PRV		;"R" SET?
	BNE	17$		;YES, MOVE IT

	TSTB	PSV		;"S" VALUE SET?
	BNE	18$		;YES, LEAVE IT

17$:	MOVB	RV,SV		;MOVE "R" STUFF TO "S"
	MOVB	PRV,PSV
	MOVB	RVAL,SVAL
	MOVB	PRVAL,PSVAL
18$:	MOVB	#6,RV		;SET "R" = D
	FLAG	PRV		;FLAG IT
	TSTB	SV		;"S" SET FOR B?
	BNE	20$		;NO

	MOVB	#4,SV		;YES, SWITCH TO A
	BR	20$		;CARRY EXIT

20$:	CMPB	FUNC,#2		;"SUBS"?
	BNE	10$		;NO

	MOVB	#1,FUNC		;NO, SWITCH TO "SUBR"
	BR	10$
�
;	JSR	PC,PRCRY	;PROCESS CARRY FIELD

PRCRY::
	TSTB	PCRRY		;"-" ALLREADY FOUND?
	BNE	62$		;YES

63$:	FIELD	14$,14$,5$,1$,7$
	RTS	PC		;ERROR EXIT

1$:	CMP	R1,#',		;COMMA?
	BNE	61$		;NO

	FLAG	PEQUF		;YES, FAG EQUATE
	RTS	PC

61$:	TSTB	PCRRY		;0=1ST TIME IN
	BNE	4$		;ALLREADY HAVE A + OR - SIGN

	INCB	PCRRY		;FLAG 1ST TIME THROUGH
	CMP	R1,#'+		;PLUS?
	BNE	3$		;NO

	TSTB	FUNC		;IF YES, FUNC MUST = "ADD"
	BEQ	63$		;IT'S OK, GET THE FIELD

2$:	ERR	'C		;CARRY ERROR
	RTS	PC

3$:	CMP	R1,#'-		;MINUS?
	BNE	2$		;NO = ERROR

62$:	TSTB	FUNC		;YES, CHECK FOR "ADD"
	BEQ	2$		;"ADD" = ERROR

	CMPB	FUNC,#2		;"SUBR" OR "SUBS"?
	BHI	2$		;NO = ERROR

	BR	63$		;YES = OK, GET FIELD

4$:	CMP	R1,#'(		;EXPRESSION?
	BNE	2$		;NO = ERROR

	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
	.WORD	2$		;ERROR ADDRESS

	BR	8$		;GO CHECK THE RESULTS

5$:	TSTB	PCRRY		;ALLREADY HAVE A + OR - ?
	BEQ	2$		;NO = ERROR

	JSR	R5,SRC		;SEARCH FOR "C" OR "CN"
	.WORD	FCSRC,6$	;CARRY TABLE, NOT FOUND ADDR.

	CLRB	CCONT		;"C" OR "CN" = NO CONTROL
	RTS	PC

6$:	JSR	R5,LKUP		;GET THE SYMBOL
	.WORD	2$		;ERROR

	BR	8$		;GO CHECK THE VALUE
�
7$:	TSTB	PCRRY		;ALLREADY HAVE A + OR - ?
	BEQ	2$		;NO = ERROR

	JSR	R5,EVNUM	;EVALUATE THE NUMBER
	.WORD	2$		;ERROR ADDR

8$:	TST	R1		;VALUE = 0?
	BEQ	13$		;YES

	CMP	R1,#1		;VALUE = 1?
	BNE	2$		;NO = ERROR

	TSTB	FUNC		;FUNCTION = "ADD"?
	BNE	10$		;NO

9$:	MOVB	.SEC,CCONT	;YES, SET CARRY
	FLAG	PCRY		;FLAG IT
	RTS	PC

10$:	CMPB	FUNC,#2		;FUNC = "SUBR" OR "SUBS"?
	BHI	12$		;NO = NO CONTROL

11$:	MOVB	.CLC,CCONT	;YES = CLEAR CARRY
	FLAG	PCRY		;FLAG IT
12$:	RTS	PC

13$:	TSTB	FUNC		;"ADD"?
	BEQ	11$		;YES = CLEAR CARRY

	CMPB	FUNC,#2		;"SUBR" OR "SUBS"?
	BLOS	9$		;YES = SET CARRY

	RTS	PC		;NO = NO CONTROL

14$:	TSTB	PCRRY		;1ST TIME IN?
	BNE	2$		;NO = ERROR

	TSTB	PSPEC		;SPECIAL FUNC?
	BEQ	13$		;NO, GO CHECK FUNCTION

	RTS	PC		;YES, EXIT
�
;	JSR	PC,FIGR		;FIGURE "R AND S" OUT
;  RV AND SV TYPE CODES
;  TYPE	  CODE
;    A      4
;    B      0
;    Q      2
;    Z      5
;    D      6
;  RV+SV-4=SOURCE CODE INDEX

FIGR::
	CMPB	RV,#4		;"R" = A?
	BNE	2$		;NO, GO CHECK "S"

	TSTB	PAADDR		;IS "A" USED?
	BEQ	1$		;NO, TAKE IT

	CMPB	RVAL,AADDR	;ARE "R" AND "A" SAME?
	BEQ	2$		;YES, GO CHECK "S"

60$:	ERR	'R		;TERM "R" ERROR
	RTS	PC

1$:	MOVB	RVAL,AADDR	;SET "R" IN "A" ADDR
	FLAG	PAADDR		;FLAG "A"
2$:	CMPB	SV,#4		;"S" = A?
	BEQ	3$		;YES

	TSTB	SV		;"S" = B?
	BNE	7$		;NO, GO SET SOURCE CODE

	BR	4$		;YES, CHECK B

3$:	CMPB	RV,#4		;"R" = A?
	BEQ	61$		;YES, SWITCH "S" TO B

	CMPB	RV,#5		;"R" = Z?
	BNE	5$		;NO, GO CHECK A

61$:	CLRB	SV		;SET "S" = B
4$:	TSTB	PBADDR		;B ADDR USED?
	BEQ	63$		;NO, TAKE IT

	CMPB	SVAL,BADDR	;"S" AND B THE SAME?
	BEQ	7$		;YES, GO SET SOURCE

	CMPB	RV,#5		;"R" = Z?
	BNE	6$		;NO = ERROR

	MOVB	#4,SV		;SET "S" = A
5$:	TSTB	PAADDR		;A ADDR USED?
	BEQ	62$		;NO, TAKE IT

	CMPB	SVAL,AADDR	;"S" AND A THE SAME?
	BEQ	7$		;YES, GO SET SOURCE

6$:	ERR	'S		;"S" TERM ERROR
	RTS	PC

63$:	MOVB	SVAL,BADDR	;SET B = "S"
	FLAG	PBADDR
	BR	7$

62$:	MOVB	SVAL,AADDR	;SET A = "S"
	FLAG	PAADDR
7$:	TSTB	RV		;"R"=B?
	BNE	11$		;NO

	CMPB	FUNC,#5		;"NOTAND"?
	BEQ	60$		;YES, CAN'T SWAP

	CMPB	SV,#4		;"S"=A?
	BEQ	9$		;YES, SWAP OK

	CMPB	SV,#5		;"S"=Z?
	BNE	60$		;NO = ERROR

9$:	CMPB	FUNC,#1		;"SUBR"?
	BNE	10$		;NO

	INCB	FUNC		;YES, SWITCH TO "SUBS"
	BR	11$

10$:	CMPB	FUNC,#2		;"SUBS"?
	BNE	11$		;NO

	DECB	FUNC		;YES, SWITCH TO "SUBR"
11$:	MOVB	RV,R0		;GET "R" CODE
	MOVB	SV,R1
	ADD	R1,R0		;GET "R"+"S" CODE
	SUB	#4,R0		;R+S-4 = INDEX
	MOVB	18$(R0),SORC	;SET FUNCTION SOURCE CODE
	FLAG	PSRC		;FLAG IT
	RTS	PC

18$:	.BYTE	1,3,0,2,6,4,5,7	;SOURCE CODE TABLE
�
;	JSR	PC,PREQU	;PROCESS EQUATE FIELD

PREQU::
	FIELD	62$,2$,6$,4$,7$
1$:	ERR	'E		;EQUATE ERROR
	RTS	PC		;DONE EXIT

2$:	CLRB	PEQUF		;CLEAR EQUATE FLAG
	BR	PREQU		;START OVER

3$:	CMP	R1,#',		;COMMA?
	BEQ	63$		;YES

	CMP	R1,#'#		;EQUATE FIELD?
	BNE	60$		;NO

63$:	FLAG	PEQUF		;YES, FLAG IT
	BR	PREQU		;GET THE FIELD

60$:	TSTB	PEQUF		;ENDING OR NONE?
	BEQ	62$		;NONE

	BR	1$		;ENDING = ERROR

4$:	TSTB	PEQUF		;THIS AN EQU. FIELD?
	BEQ	3$		;NO, STARTING ONE?

	BMI	3$		;YES

	CMP	R1,#'(		;EXPRESSION?
	BEQ	5$		;YES

	JSR	R5,CKPC		;IS IT THE PC?
	.WORD	8$		;YES
	BR	1$		;DONE EXIT

5$:	JSR	R5,EVEXP	;EVALUATE EXPRESSION
	.WORD	1$		;ERROR ADDRESS
	BR	8$		;GO SET THE VALUE

6$:	TSTB	PEQUF		;THIS AN EQUATE?
	BEQ	62$		;NO = COMMENT FIELD

	BMI	1$		;ERROR

	JSR	R5,LKUP		;GET SYMBOL
	.WORD	61$		;ERROR

	BR	8$		;GO SET VALUE

61$:	ERR	'U		;UNDEFINED
62$:	RTS	PC

7$:	TSTB	PEQUF		;THIS AN EQUATE?
	BEQ	62$		;NO = COMMENT FIELD

	BMI	1$		;ERROR

	JSR	R5,EVNUM	;EVALUATE THE NUMBER
	.WORD 	1$		;ERROR ADDRESS

8$:	MOV	#INSTR,R3	;INSTRUCTION POINTER
	BIS	R1,(R3)+	;LSB
	ASR	R0		;WORD COUNT
	BR	10$

9$:	BIS	(R2)+,(R3)+	;SET INSTRUCTION
10$:	DEC	R0
	BNE	9$

	NEGB	PEQUF		;SET CONTROL FIELD DONE
	BR	PREQU		;START OVER
�
;	JSR	R5,PRADD	;PROCESS JUMP ADDRESS FIELD
;	.WORD	COMMENT EXIT

PRADD::
	TSTB	PJADDR		;ADDRESS FIELD NEEDED?
	BEQ	9$		;NO

1$:	FIELD	2$,1$,6$,5$,7$
2$:	ERR	'J		;JUMP ADDRESS ERROR
3$:	MOV	(R5),R5		;COMMENT EXIT
	RTS	R5

5$:	JSR	R5,CKPC		;IS IT THE PC?	
	.WORD	8$		;YES
	CMP	R1,#'(		;EXPRESSION?
	BNE	2$		;NO = ERROR

	JSR	R5,EVEXP	;EVALUATE EXPRESSION
	.WORD	3$		;ERROR ADDR.
	BR	8$		;GO SET VALUE

6$:	JSR	R5,SRC		;LOOK FOR "D"
	.WORD	F3SRC,10$	;TABLE, NOT FND
	BR	9$		;NO ADDR IF = "D"

10$:	JSR	R5,LKUP		;LOOK FOR SYMBOL
	.WORD	2$		;ERROR ADDR.
	BR	8$		;SET VALUE

7$:	JSR	R5,EVNUM	;EVALUATE NUMBER
	.WORD	2$		;ERROR ADDR.
8$:	MOV	R1,IDTA		;SET ADDRESS
	MOVB	.IMC,IMC	;SET IMMEDIATE CONTROL CODE
9$:	MOVB	#-1,PEQUF	;SET AS END OF CONTROL
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5
�
;  R1 = START OF FIELD BYTE POINTER
;  R2 = END OF FIELD+1
;	JSR	R5,SRC		;SEARCH FOR ENTRY MATCH
;	.WORD	SEARCH TABLE ADDRESS
;	.WORD	NOT FOUND EXIT ADDRESS (R1, R2 SAVED)
;  R1 = ENTRY VALUE IF FOUND

SRC::
	MOV	(R5)+,R0	;GET SEARCH TABLE ADDR.
	MOV	R2,-(SP)	;SAVE END POINTER
	SUB	R1,R2		;R2 = BYTE COUNT

1$:	MOV	R1,R3		;R3 = START OF FIELD
	MOVB	(R0)+,R4	;BYTE COUNT FROM TABLE
	BEQ	5$		;NOT FOUND EXIT

	CMP	R2,R4		;LENGTH =?
	BNE	3$		;NO = A MISS

2$:	CMPB	(R3)+,(R0)+	;CHECK A BYTE
	BNE	4$		;GOT A MISS

	DEC	R4		;DONE?
	BNE	2$		;NO

	MOVB	(R0),R1		;HIT! GET THE VALUE
	BIC	#177400,R1	;CLEAR LEFT BYTE (MAX=255)
	MOV	(SP)+,R2	;RESTORE END OF FIELD POINTER
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5

3$:	INC	R0		;ADJUST FOR FOLLOWING "ADD"
4$:	ADD	R4,R0		;MOVE TO NEXT ENTRY
	BR	1$		;CHECK IT

5$:	MOV	(SP)+,R2	;RESTORE END POINTER
	MOV	(R5),R5		;NOT FOUND EXIT
	RTS	R5
�




1280_Microcode/asm1.mac
; 2ND HALF OF ASM.MAC

	.PSECT	ASM

;	JSR	PC,ISCAN	;INIT SCANNER
ISCAN::
	MOV	#LINE,SPTR	;RESET SCAN POINTER
	RTS	PC


;	JSR	R5,SCAN		;SCAN FOR A FIELD
;  R2 AND SPTR = FIELD END+1
;  R1 = FIELD START
;  R1 = ASCII CHARACTER IF FIELD TYPE IS 3.
;  R0 = FIELD TYPE
;  PARAMETERS ARE RETURN ADDRESS POINTERS
;	.WORD	ADDR0	;0 END OF LINE
;	.WORD	ADDR1	;1 SPACE FIELD
;	.WORD	ADDR2	;2 SYMBOL FIELD
;	.WORD	ADDR3	;3 SPECIAL CHARACTER
;	.WORD	ADDR4	;4 NUMBER FIELD
;  IN-LINE RETURN IF NUMBER FIELD ERROR DETECTED

SCAN:
	CLRB	SOCTF		;OCTAL FLAG
	CLRB	SDECF		;DECIMAL FLAG
	CLRB	SHEXF		;HEX. FLAG
	CLRB	SERRF		;ERROR FLAG
	MOV	SPTR,R2		;START OF FIELD
	MOV	R2,-(SP)	;SAVE START
	MOVB	(R2)+,R1	;GET A BYTE
	BPL	2$		;BR IF NOT END OF FIELD

	CLR	R0		;TYPE = END OF LINE
1$:	DEC	R2		;BACKUP
63$:	MOV	R2,SPTR		;SAVE FIELD POINTER
	MOV	(SP)+,R1	;START OF FIELD POINTER
	ASL	R0		;FROM WORD TO BYTE OFFSET
	ADD	R0,R5		;EXIT POINTER
	MOV	(R5),R5		;EXIT ADDRESS
	RTS	R5		;EXIT

2$:	CMP	R1,#40		;SPACE?
	BEQ	3$		;YES
	CMP	R1,#11		;OR TAB?
	BNE	4$		;NO
3$:	MOVB	(R2)+,R1	;GET NEXT BYTE
	CMP	R1,#40		;STILL SPACE FIELD?
	BEQ	3$		;YES
	CMP	R1,#11		;OR TAB?
	BEQ	3$		;YES
	MOV	#1,R0		;TYPE = SPACE FIELD
	BR	1$		;EXIT

4$:	CMP	R1,#47		;APOSTR.?
	BNE	5$		;NO
	INCB	SOCTF		;FLAG IT AS OCTAL
	BR	6$		;GET NEXT BYTE

5$:	CMP	R1,#'0		;NUMBER?
	BLO	14$		;NO
	CMP	R1,#'9		;NUMBER?
	BHI	14$		;NO
	BR	62$		;CHECK DEC.
�
6$:	MOVB	(R2)+,R1	;GET A BYTE
62$:	CMP	R1,#'0		;NUMBER?
	BLO	9$		;NO
	CMP	R1,#'7		;OCTAL?
	BLOS	6$		;YES
	CMP	R1,#'9		;DECIMAL?
	BHI	7$		;NO
	INCB	SDECF		;FLAG IT AS DECIMAL
	BR	6$

7$:	CMP	R1,#'A		;HEX?
	BLO	9$		;NO

	CMP	R1,#'F		;HEX?
	BHI	8$		;NO

	INCB	SHEXF		;FLAG IT AS HEX.
	BR	6$

8$:	CMP	R1,#'Z		;LETTER?
	BHI	9$		;NO

	INCB	SERRF		;YES = ERROR
	BR	6$

9$:	TSTB	SERRF		;ERROR?
	BNE	12$		;YES

	CMP	R1,#'.		;DECIMAL POINT?
	BNE	11$		;NO

	INC	R2		;ADJUST TO MISS "."
	INCB	SDECF		;SET DEC. FLAG
	TSTB	SOCTF		;OCTAL?
	BNE	12$		;YES = ERROR

	TSTB	SHEXF		;HEX?
	BNE	12$		;YES = ERROR

10$:	MOV	#4,R0		;TYPE = NUMBER
	BR	1$

11$:	TSTB	SOCTF		;OCTAL?
	BEQ	13$		;NO

	TSTB	SDECF		;DECIMAL?
	BNE	12$		;YES = ERROR

	TSTB	SHEXF		;HEX?
	BEQ	10$		;NO = OK

12$:	DEC	R2		;MOVE BACK TO BAD CHAR.
	MOV	R2,SPTR		;SAVE FIELD POINTER
	MOV	(SP)+,R1	;START OF FIELD
	ADD	#10.,R5		;IN-LINE EXIT = NUMBER ERROR
	ERR	'N		;NUMBER ERROR
	RTS	R5

13$:	CLRB	SDECF		;CLEAR DECIMAL FLAG
	INCB	SHEXF		;AND SET HEX FLAG
	BR	10$		;TAKE OK EXIT
�
14$:	CMP	R1,#'$		;SYMBOL?
	BEQ	15$		;YES

	CMP	R1,#'.		;SYMBOL?
	BEQ	15$		;YES

	CMP	R1,#'A		;SYMBOL?
	BLO	17$		;NO

	CMP	R1,#'Z		;SYMBOL?
	BHI	17$		;NO

15$:	MOVB	(R2)+,R1	;GET NEXT BYTE
	CMP	R1,#'$		;SYMBOL?
	BEQ	15$		;YES

	CMP	R1,#'.		;SYMBOL?
	BEQ	15$		;YES

	CMP	R1,#'Z		;SYMBOL?
	BHI	16$		;NO

	CMP	R1,#'A		;SYMBOL?
	BHIS	15$		;YES

	CMP	R1,#'9		;SYMBOL?
	BHI	16$		;NO

	CMP	R1,#'0		;SYMBOL?
	BHIS	15$		;YES

16$:	MOV	R2,R0
	SUB	(SP),R0		;FIELD LENGTH+1
	CMP	R0,#2		;LENGTH = 1?
	BNE	18$		;NO

	CMPB	@(SP),#'.	;CHAR. = PC?
	BNE	18$		;NO

	DEC	R2		;YES, BACKUP
	MOVB	@(SP),R1	;GET "."
17$:	MOV	#3,R0		;TYPE = SPECIAL CHARACTER
	MOV	R1,(SP)		;REPLACE START POINTER WITH ASCII
	JMP	63$

18$:	MOV	#2,R0		;TYPE = SYMBOL
	JMP	1$
�
;	JSR	R5,GARG		;GET AN ARGUMENT
;	.WORD	ERROR OR NOT FOUND EXIT ADDRESS
;  R0 = 0 IF ARG. IS 16 BIT LOCAL SYMBOL
;  R0 = 2,4,6,8 (BYTE COUNT OF ARGUMENT VALUE)
;  R1 = LSB 16 BITS OF ARG. VALUE
;  R2 = VALUE ADDRESS POINTER +2 (2ND LSB 16 BITS)

GARG::
	FIELD	1$,GARG,LKUP,2$,EVNUM
1$:	MOV	(R5),R5		;ERROR EXIT
	RTS	R5

2$:	CMP	R1,#'(		;EXPRESSION?
	BEQ	3$		;YES
	CMP	R1,#',		;COMMA?
	BEQ	GARG		;YES, SKIP IT
	JSR	R5,CKPC		;IS IT THE PC?
	.WORD	4$		;YES

	BR	1$		;NO = ERROR

3$:	JMP	EVEXP		;EVALUATE THE EXPRESSION
4$:	TST	(R5)+		;IN-LINE EXIT
	RTS	R5



;  CHECK DECIMAL POINT FOR =PC
;  R1 = ASCII SPECIAL CHARACTER FROM "SCAN"
;	JSR	R5,CKPC		;CHECK FOR "=PC"
;	.WORD	EXIT ADDR. IF DECIMAL POINT IS THE PC
;  IF YES THEN R0 = 2 = 16 BIT TYPE
;		R1 = 16 BIT VALUE ( = LOC)
;		R2 = VALUE+2 POINTER  ( LOC+2)
CKPC::
	CMP	R1,#'.		;PC?
	BEQ	1$		;YES

	TST	(R5)+		;IN-LINE EXIT = "NOT PC"
	RTS	R5

1$:	MOV	#2,R0		;PC TYPE = 2
	MOV	#LOC,R2
	MOV	(R2)+,R1	;SET R1 AND R2
	MOV	(R5),R5		;"=PC" EXIT
	RTS	R5
�
;  R0	WORK REGISTER
;  R1	ADDR. OF NEXT INPUT CHARACTER
;  R2	MAX. # OF INPUT CHARS. LEFT
;  R3	SAVED
;  R4	RAD50 VALUE TO BE RETURNED
;  R5	SAVED
;	JSR	PC,R50WD	;CONVERT ASCII TO RAD50

R50WD::
	MOV	R5,-(SP)
	MOV	#2,-(SP)	;SET LOOP COUNT (3 CHAR.)
	CLR	R4		;ZERO ACCUMULATOR WORD
1$:	CLR	R0		;ZERO CURRENT RAD50 CHAR.
	DEC	R2		;ANY CHARS LEFT TO PROCESS?
	BMI	5$		;NO

	MOV	#6$,R5		;CONVERSION TABLE POINTER
	BISB	(R1)+,R0	;GET NEXT INPUT CHAR.
2$:	CMPB	(R5)+,R0	;.LE. UPPER LIMIT?
	BLO	3$		;NO

	CMPB	(R5)+,R0	;.GT. LOWER LIMIT?
	BLO	4$		;YES, FOUND CONVERSION

	DEC	R5		;ELSE BUMP OVER REMAINDER
3$:	ADD	#3,R5		;  OF TABLE ENTRY
	TST	(R5)		;END OF TABLE?
	BNE	2$		;NO

	NEG	R2		;YES, SET FOR NON-RAD50
	CLR	R0
4$:	ADD	(R5),R0		;DO TRANSLATION
5$:	ASL	R4		;MULTIPLY PREVIOUS BY 50
	ASL	R4
	ASL	R4		;*8
	ADD	R4,R0
	ASL	R4		;*16
	ASL	R4		;*32
	ADD	R0,R4		;ADD IN CURRENT CHAR.
	DEC	(SP)		;ANY CHAR. LEFT?
	BPL	1$		;YES

	TST	(SP)+		;POP LOOP COUNT
	MOV	(SP)+,R5
	RTS	PC		;DONE

6$:	.BYTE	132,100		;A-Z
	.WORD	-100
	.BYTE	71,57		;0-9
	.WORD	-22
	.BYTE	40,37		;SPACE
	.WORD	-40
	.BYTE	44,43		;$
	.WORD	-11
	.BYTE	56,55		;PERIOD
	.WORD	-22
	.WORD	0		;END OF TABLE
�
;  R1 = START OF FIELD
;  R2 = END OF FIELD+1
;	JSR	R5,LKUP		;SYMBOL LOOKUP
;	.WORD	NOT FOUND EXIT ADDRESS
;  R0 = -1 IF LOCAL SECTION NOT FOUND
;  R0 = 0 IF LOCAL SYMBOL FOUND
;  R0 = 2 IF 16 BIT SYMBOL FOUND
;  R0 = 4 IF 32 BIT SYMBOL FOUND
;  R0 = 6 IF 48 BIT SYMBOL FOUND
;  R0 = 8 IF 64 BIT SYMBOL FOUND
;  R1 = LSB 16 BITS OF SYMBOL VALUE
;  R2 = SYMBOL VALUE ADDRESS+2 IF SYMBOL FOUND

;  SYMBOL CONVERTED TO RAD50 SAVED IN "SSYM" (2 WORDS)
;  IF PERM. SYMBOL NOT FOUND THEN:
;	R0 = 2
;	R1 = SYMBOL LENGTH
;	R2 = NEXT FIELD POINTER
;	R3,R4 = RAD50 SYMBOL

LKUP::
	SUB	R1,R2		;FIELD BYTE COUNT
	MOV	R2,-(SP)	;SAVE BYTE COUNT
	MOVB	(R1),-(SP)	;GET FIRST CHAR.
	CMP	R2,#6		;FIELD > 6 CHARS.?
	BLOS	1$		;NO

	MOV	#6,R2		;YES, TRUNCATE TO 6.
1$:	JSR	PC,R50WD	;CONVERT 1ST 3 TO RAD50

	MOV	R4,SSYM		;SAVE 1ST RAD50 WORD
	JSR	PC,R50WD	;CONVERT NEXT 3

	MOV	R4,SSYM+2	;SAVE 2ND WORD
	MOV	SSYM,R3		;R3,R4 = RAD50 SYMBOL
	CLR	R0		;0 = LOCAL SYMBOL
	CMP	(SP)+,#'$	;LOCAL?
	BNE	10$		;NO, MUST BE PERM.

;  SEARCH FOR LOCAL SYMBOL
	TST	(SP)+		;POP THE LENGTH
	MOV	SYMLS,R2	;GET LOCAL SECTION POINTER
2$:	CMP	(R2),JSTMD	;CURRENT SECTION?
	BHIS	3$		;YES, LOOK FOR THE SYMBOL

	MOV	-(R2),R2	;ADVANCE TO THE NEXT SECTION
	BR	2$		;CHECK IT

3$:	BHI	8$		;END = NOT FOUND

	TST	-(R2)
	MOV	R2,R1		;SAVE PTR TO NEXT SECTION PTR
4$:	CMP	(R2),R3		;HIT?
	BEQ	6$		;YES

	TST	-(R2)		;NO, MOVE TO NEXT ENTRY
	TST	-(R2)		;DONE?
	BNE	4$		;NO

5$:	MOV	(R5),R5		;NOT FOUND EXIT
	RTS	R5
�
6$:	MOV	-(R2),R1	;GET SYMBOL VALUE
7$:	BPL	57$		;NO ERROR

	MOV	R0,-(SP)
	ERR	'M		;MULT. DEF.
	MOV	(SP)+,R0
57$:	BIC	#177600,R0	;CLEAR ERROR BIT
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5

8$:	DEC	R0		;FLAG "SECTION NOT FOUND"
	BR	5$		;NOT FOUND EXIT

;  PERMANENT SYMBOL SEARCH
10$:	TST	(R0)+		;SET TYPE = 2 (16 BIT VALUE)
	MOV	SYMS,R2		;INIT ENTRY POINTER
11$:	TST	(R2)		;END OF TABLE?
	BEQ	17$		;YES = NOT FOUND

	CMP	(R2)+,R3	;1ST WORD OK?
	BNE	12$		;NO

	CMP	(R2)+,R4	;2ND WORD OK?
	BNE	13$		;NO

	CMPB	(R2)+,(SP)	;LENGTH OK?
	BNE	14$		;NO

;  GOT A HIT!
	TST	(SP)+		;POP
	MOVB	(R2)+,R0	;R0 = TYPE
	MOV	(R2)+,R1	;SET LSB 16 BITS OF VALUE
	TSTB	R0		;ERROR?
	BR	7$		;TAKE IN-LINE EXIT

12$:	TST	(R2)+		;MOVE TO 2ND WORD
13$:	TSTB	(R2)+		;MOVE TO ENTRY LENGTH
14$:	MOVB	(R2)+,R1	;GET OFFSET TO NEXT ENTRY
	BMI	16$		;ERROR FLAG ON

15$:	ADD	R1,R2		;MOVE TO NEXT ENTRY
	BR	11$		;GO CHECK IT

16$:	BIC	#177600,R1	;REMOVE ERROR FLAG
	BR	15$

17$:	MOV	(SP)+,R1	;SYMBOL LENGTH
	BR	5$		;NOT FOUND EXIT
�
;	CLR	PLBLF		;ENTER WITH FLAG = 0
;  R1 = START OF LABEL FIELD
;  R2 = END OF FIELD+1
;	JSR	PC,SLBL		;ENTER LABEL IN SYMBOL TABLE
;  "VADD" = ADDR OF SYMBOL ENTRY VALUE

SLBL::
	TSTB	COND		;CODE ON?
	BNE	61$		;NO

	JSR	R5,LKUP		;LOOK FOR THE LABEL
	.WORD	1$		;NOT FOUND

;  FOUND THE LABEL
	MOVB	R0,LTYP		;SAVE THE SYMBOL TYPE
	BEQ	63$		;DON'T MOVE BACK IF LOCAL

	TST	-(R2)		;MOVE BACK TO ENTRY VALUE
63$:	MOV	R2,VADD		;SAVE VALUE ADDRESS
	TSTB	PASS		;PASS 1?
	BNE	2$		;NO = OK

	MOV	R2,-(SP)
	TST	R0		;LOCAL?
	BEQ	62$		;YES

	TST	-(R2)		;MOVE TO ERROR FLAG
62$:	BIS	#100000,(R2)	;FLAG "M" ERROR
	MOV	(SP)+,R2
	BR	2$

1$:	CMP	R0,#0		;LOCAL?
	BLE	4$		;YES

	MOV	R3,(R2)+	;ENTER SYMBOL (RAD50 WORD)
	MOV	R4,(R2)+
	SWAB	R0		;POSITION THE TYPE BYTE
	BIS	R0,R1		;ADD TYPE TO LENGTH
	MOV	R1,(R2)+	;ENTER IT IN SYMBOL TABLE
	MOV	R2,VADD		;SET VALUE ADDRESS POINTER
	MOV	LOC,(R2)+	;SET VALUE = CURRENT PC
	MOVB	#2,LTYP		;SET TYPE = 16 BIT
	CLR	(R2)		;MARK END OF SYMBOL TABLE
	INCB	PLBLF		;FLAG THE ENTRY
2$:	TSTB	LTYP		;LOCAL?
	BLE	3$		;YES

	MOV	JSTMD,OLDMD	;SAVE CURRENT LOCAL SECTION PTR
	MOV	LOC,JSTMD	;SET LOCAL SECTION PTR TO CURRENT PC
3$:	INCB	PLBLF		;1=OLD LABEL, 2=NEW LABEL
61$:	RTS	PC

4$:	BEQ	6$		;GO DO LOCAL SYMBOL
�
;  CREATE A LOCAL SECTION
	CMP	(R2),#77777	;END OF LOCAL TABLE?
	BEQ	5$		;YES

	ERR	'X		;ASSEMBLER ERROR
	RTS	PC

5$:	MOV	JSTMD,(R2)	;SET CURRENT SECTION
	TST	-(R2)
	MOV	R2,R1		;R1 PTS TO NEXT SECTION PTR
	TST	-(R2)		;R2 = END OF LOCAL SECTION

;  R1 POINTS TO NEXT SECTION PTR
;  R2 POINTS TO END OF LOCAL SECTION
6$:	CLRB	LTYP		;SET TYPE = LOCAL
	MOV	SSYM,(R2)	;ENTER RAD50 NAME
	TST	-(R2)
	MOV	R2,VADD		;SAVE PTR TO VALUE
	MOV	LOC,(R2)	;USE CURRENT PC AS VALUE
	CLR	-(R2)		;MARK END OF LOCAL SECTION
	TST	-(R2)
	MOV	R2,(R1)		;UPDATE NEXT SECTION PTR
	MOV	#77777,(R2)	;FLAG AS END OF TABLE
	BR	3$		;TAKE LOCAL EXIT
�
;  R0 = 4  (FIELD TYPE FOR A NUMBER)
;  R1 = START OF FIELD POINTER
;  R2 = END OF FIELD+1 POINTER (SAVED IN SPTR)
;  ENTER WITH NUMBER FLAGS SET. (SOCTF, SDECF, SHEXF)
;	JSR	R5,EVNUM	;EVALUATE A NUMBER
;	.WORD	ERROR EXIT ADDRESS
;  R0 = VALUE TYPE (BYTE COUNT) OF 2,4,6 OR 8.
;  R1 = LSB 16 BITS OF NUMBER VALUE
;  R2 = POINTER TO VALUE ADDRESS+2

EVNUM::
	TSTB	SOCTF		;OCTAL?
	BEQ	1$		;NO

	INC	R1		;YES, SKIP THE "'"
1$:	TSTB	SDECF		;DECIMAL?
	BEQ	2$		;NO

	DEC	R2		;YES, SKIP THE "."
2$:	MOV	#RSLT,-(SP)	;EVALUATION RESULT STORAGE AREA
	MOV	(SP),R3
	CLR	(R3)+		;INIT THE RESULT TO 0
	CLR	(R3)+
	CLR	(R3)+
	CLR	(R3)+
3$:	MOV	(SP),R3		;INIT RESULT POINTER
	MOV	#4,R4		;HEX SHIFT COUNT
	TSTB	SHEXF		;HEX?
	BNE	4$		;YES

	TSTB	SDECF		;NOT HEX, IS IT DEC.?
	BNE	11$		;YES

	DEC	R4		;MUST BE OCTAL, ADJUST SHIFT CNT
4$:	JSR	PC,9$		;SHIFT RESULT LEFT

5$:	MOVB	(R1)+,R0	;GET A DIGIT
	BIC	#177600,R0	;STRIP IT
	SUB	#'9,R0		;DECIMAL?
	BLE	6$		;YES
	SUB	#'A-'9-1,R0	;NO, ADJUST FOR HEX
6$:	ADD	#9.,R0		;R0 = VALUE OF DIGIT
	MOV	(SP),R3		;INIT RESULT POINTER
	ADD	R0,(R3)+	;ADD DIGIT VALUE TO RESULT
	ADC	(R3)+
	ADC	(R3)+
	ADC	(R3)+
	CMP	R1,R2		;DONE?
	BLT	3$		;NO, DO NEXT

;  DONE, NOW FIGURE VALUE TYPE
	MOV	#3,R1		;LOOP COUNTER
	MOV	#8.,R0		;MAX BYTE COUNT
7$:	TST	-(R3)		;ANY VALUE HERE?
	BNE	8$		;YES, QUIT

	TST	-(R0)		;NO, REDUCE BYTE COUNT
	DEC	R1		;DONE LOOKING?
	BNE	7$		;NO

8$:	MOV	(SP)+,R2	;R2 POINTS TO RESULT
	MOV	(R2)+,R1	;R1=LSB OF VALUE, R2=ADDR+2
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5
�
SHIFTL	= .
9$:	MOV	R3,R0		;RESULT POINTER
	TST	R4		;ANY SHIFTS?
	BNE	10$		;YES

	ADD	#6,R3		;ADVANCE R3
	RTS	PC

10$:	MOV	R0,R3		;INIT RESULT POINTER
	CLC
	ROL	(R3)+		;SHIFT 1 BIT LEFT
	ROL	(R3)+
	ROL	(R3)+
	ROL	(R3)
	DEC	R4		;DONE SHIFTING?
	BNE	10$		;NO, DO IT AGAIN
12$:	RTS	PC		;YES, EXIT

11$:	MOV	#1,R4		;READY TO SHIFT ONCE
	JSR	PC,9$		;SHIFT LEFT

	MOV	#RSLT+6,R3
	TST	-(SP)		;SAVE SPACE FOR "ADD4"
	MOV	(R3),-(SP)	;SAVE RESULT*2
	MOV	-(R3),-(SP)
	MOV	-(R3),-(SP)
	MOV	-(R3),-(SP)	;LSB
	MOV	#2,R4
	JSR	PC,9$		;SHIFT 2 (RESULT * 8.)

	MOV	#RSLT,R3
	JSR	PC,ADD4		;*2 PLUS *8 = *10

	BR	5$		;NOW ADD DIGIT VALUE


;	TST	-(SP)		;MAKE ROOM FOR "ADD4"
;	MOV	4 WORDS TO STACK
;	JSR	PC,ADD4		;ADD FROM STACK TO (R3)
;  5 WORDS ARE POPPED
;  R3 POINTS TO 4TH DESTINATION WORD
ADD4::
	MOV	(SP)+,8.(SP)	;SET RETURN IN STACK
	ADD	(SP)+,(R3)+	;*8 PLUS *2 = *10
	ADC	(R3)+
	ADC	(R3)+
	ADC	(R3)
	CMP	-(R3),-(R3)	;MOVE BACK TO 2ND WORD
	ADD	(SP)+,(R3)+
	ADC	(R3)+
	ADC	(R3)
	TST	-(R3)		;MOVE BACK TO 3RD WORD
	ADD	(SP)+,(R3)+
	ADC	(R3)
	ADD	(SP)+,(R3)	;4TH WORD
	RTS	PC
�
;  EVALUATE AN EXPRESSION
;  R2 = POINTER TO 1ST BYTE AFTER "("
;	JSR	R5,EVEXP	;EVALUATE THE EXPRESSION
;	.WORD	ERROR EXIT ADDRESS
;  R0 = EXPRESSION TYPE. (BYTE COUNT)
;  R1 = LSB 16 BIT VALUE
;  R2 POINTS TO VALUE ADDRESS +2
;  "SCAN POINTER (SPTR)" WILL BE NEXT AFTER ")"

EVEXP::
	CLR	-(SP)		;INIT RESULT STORAGE
	CLR	-(SP)
	CLR	-(SP)
	CLR	-(SP)
	MOV	#1,-(SP)	;INIT "OP TO DO" TO "+"
1$:	FIELD	2$,2$,4$,6$,5$

2$:	ERR	'W		;EXPRESSION ERROR
3$:	ADD	#10.,SP		;ERROR, POP EVERYTHING
	MOV	(R5),R5		;ERROR EXIT
	RTS	R5

4$:	JSR	R5,LKUP		;GET SYMBOL VALUE
	.WORD	2$		;ERROR
	BR	8$		;GO DO "OP TO DO"

5$:	JSR	R5,EVNUM	;EVALUATE NUMBER
	.WORD	2$		;ERROR
	BR	8$		;GO DO "OP TO DO"

6$:	JSR	R5,CKPC		;IS IT "."?
	.WORD	8$		;YES
	MOV	#63$,R0		;OP TABLE POINTER
7$:	TSTB	(R0)		;END OF TABLE?
	BEQ	2$		;YES = ERROR

	CMPB	(R0)+,R1	;OP MATCH?
	BNE	7$		;NO, LOOK AGAIN

	SUB	#63$+1,R0	;OP # = "("?
	BNE	20$		;NO

	JSR	R5,EVEXP	;YES, EVALUATE EXPRESSION
	.WORD	3$		;ERROR

8$:	MOV	(SP),R3		;"OP TO DO"?
	BEQ	2$		;NO = ERROR

	DEC	R3
	ASL	R3		;MAKE BYTE OFFSET
	TST	-(R2)		;MOVE R2 BACK TO VALUE
	MOV	#62$+6,R4	;WORK SPACE
	CMP	R2,#62$		;SOURCE=DESTINATION?
	BEQ	59$		;YES

	CLR	(R4)		;INIT WORK SPACE TO 0
	CLR	-(R4)
	CLR	-(R4)
	CLR	-(R4)
	ASR	R0		;MAKE WORD COUNT
�
9$:	MOV	(R2)+,(R4)+	;MOVE VALUE TO WORK SPACE
	DEC	R0		;DONE?
	BNE	9$		;NO
59$:	JMP	@10$(R3)	;GO DO "OP TO DO"

10$:	.WORD	11$,13$,16$,17$,14$	;+-*/^

11$:	MOV	#62$+6,R0	;VALUE MSB POINTER
	MOV	SP,R3
	TST	(R3)+		;ADVANCE TO RESULT LSB
	TST	-(SP)		;SPACE FOR "ADD4" RETURN
	MOV	(R0),-(SP)	;PUSH MSB 1ST
	MOV	-(R0),-(SP)
	MOV	-(R0),-(SP)
	MOV	-(R0),-(SP)	;PUSH LSB LAST
	JSR	PC,ADD4		;ADD (SP),(R3)

12$:	CLR	(SP)		;CLEAR "OP TO DO"
	BR	1$		;GET NEXT "OP TO DO"

;  SUBTRACT
13$:	MOV	#62$,R3		;LSB OF VALUE
	MOV	SP,R4
	TST	(R4)+		;LSB OF RESULT
	SUB	(R3)+,(R4)+	;SUBTRACT
	SBC	(R4)+
	SBC	(R4)+
	SBC	(R4)
	CMP	-(R4),-(R4)	;MOVE BACK TO 2ND WORD
	SUB	(R3)+,(R4)+
	SBC	(R4)+
	SBC	(R4)
	TST	-(R4)		;MOVE BACK TO 3RD WORD
	SUB	(R3)+,(R4)+	;SUBTRACK
	SBC	(R4)
	SUB	(R3),(R4)
	BR	12$

;  SHIFT
14$:	MOV	#62$+6,R3	;MSB OF VALUE
	TST	(R3)		;=0?
	BNE	2$		;NO = ERROR

	TST	-(R3)		;3RD WORD = 0?
	BNE	2$		;NO = ERROR

	TST	-(R3)		;2ND WORD = 0?
	BNE	2$		;NO = ERROR

	CMP	-(R3),#64.	;LSB REASONABLE?
	BHI	2$		;NO = ERROR

	MOV	(R3),R4		;SHIFT COUNT
	MOV	SP,R3
	TST	(R3)+		;RESULT POINTER

	JSR	PC,SHIFTL	;GO SHIFT RESULT LEFT

	BR	12$
�
;  MULTIPLY
16$:
;  DIVIDE
17$:	BR	2$		;ERROR

20$:	TST	(SP)		;"OP TO DO"?
	BNE	2$		;YES = ERROR

	DEC	R0		;")"? (END?)
	BEQ	21$		;YES

	MOV	R0,(SP)		;SET "OP TO DO"
	BR	1$		;GET NEXT VALUE

21$:	TST	(SP)+		;POP "OP TO DO"
	MOV	#8.,R0		;BYTE COUNT
	MOV	#62$,R2		;VALUE STORAGE
	MOV	(SP)+,(R2)+	;MSB
	MOV	(SP)+,(R2)+
	MOV	(SP)+,(R2)+
	MOV	(SP)+,(R2)+	;LSB
22$:	TST	-(R2)		;ANYTHING THERE?
	BNE	23$		;YES

	SUB	#2,R0		;NO, DONE?
	BNE	22$		;NO

	TST	(R0)+		;NOTHING THERE, SET = 16 BITS
23$:	MOV	#62$,R2		;LSB POINTER
	MOV	(R2)+,R1	;R1=LSB, R2 =VALUE+2
	TST	(R5)+		;IN-LINE EXIT
	RTS	R5

62$:	.WORD	0,0,0,0		;VALUE STORAGE
63$:	.ASCII	'()+-*/^'<0>	;OPERATIONS
	.EVEN
�
;	JSR	PC,DOBJ		;OUTPUT THE LAST STATEMENT
;  R1,R2 SAVED

DOBJ::
	MOV	R1,-(SP)
	MOV	R2,-(SP)
	MOV	R5,-(SP)
	TSTB	PASS		;PASS 1?
	BEQ	2$		;YES

	TSTB	PSUDOF		;ANY OBJECT?
	BNE	1$		;NO

	TSTB	COND		;CODE ON?
	BNE	1$		;NO

	JSR	PC,BLD		;BUILD THE INSTRUCTION

	JSR	PC,PUTBIN	;OUTPUT THE BINARY

1$:	JSR	PC,BLST		;BUILD INSTR. LISTING

2$:	TSTB	PSUDOF		;PSUEDO OP?
	BNE	3$		;YES

	TSTB	COND		;CODE ON?
	BNE	3$		;NO

	INC	LOC		;NO, ADVANCE THE PC
3$:	JSR	PC,UNBLD	;CLEAR FIELDS

	MOV	(SP)+,R5
	MOV	(SP)+,R2
	MOV	(SP)+,R1
	RTS	PC


;  BUILD THE INSTRUCION
BLD::
	MOV	R5,-(SP)
	MOVB	.BRC,R4		;BR. CNTRL FIELD
	MOVB	BCONT,R5	;BRANCH CONTROL DATA
	JSR	PC,INSERT	;INSERT FIELD INTO INSTRUCTION

	MOVB	.ICF,R4		;IMMEDIATE CONTROL FIELD
	MOVB	IMC,R5		;IMMEDIATE CONTROL CODE
	JSR	PC,INSERT	;INSERT IT INTO INSTR

	MOVB	.BCN,R2		;WORD BIT LENGTH
	ASR	R2		;/2
	ASR	R2		;/4
	ASR	R2		;/8 = BYTE LENGTH
	ADD	#INSTR,R2	;MSB+2 OF INSTRUCTION
	MOVB	FUNC,R0		;ALU FUNCTION
	ASL	R0
	ASL	R0
	ASL	R0
	BISB	SORC,R0		;INCLUDE ALU SOURCE
	ASL	R0
	ASL	R0
	BISB	DEST,R0		;INCLUDE ALU DESTINATION
	SWAB	R0
	BISB	BADDR,R0	;INCLUDE "B" ADDRESS
	MOVB	AADDR,R1
	ASL	R1
	ASL	R1
	ASL	R1
	ASL	R1
	BIS	R1,R0		;INCLUDE "A" ADDRESS
	SWAB	R0
	BISB	R0,-(R2)	;1ST INSTRUCTION BYTE
	SWAB	R0
	BISB	R0,-(R2)	;NEXT BYTE
	MOV	IDTA,R0		;IMMEDIATE FIELD
	BIC	IDTAM,R0	;ISOLATE FIELD
	SWAB	R0
	BISB	R0,-(R2)	;PUT IT IN
	SWAB	R0
	BISB	R0,-(R2)	;NEXT ONE IN
	MOVB	SCONT,R0	;GET SEQUENCE CONTROL
	TSTB	PFUNC		;WAS THERE A FUNCTION?
	BEQ	10$		;NO, SKIP SEQU. DEFAULT

	TSTB	PSCONT		;IS THERE SEQ. CONTRL?
	BNE	10$		;YES

	MOVB	.SEQ,R0		;NO, USE THE DEFAULT
10$:	ASL	R0
	ASL	R0
	BISB	CCONT,R0	;INCLUDE CARRY CONTROL
	ASL	R0
	BISB	SHIFT,R0	;INCLUDE SHIFT CONTROL
	SWAB	R0
	CLC
	ROL	R0
	SWAB	R0
	BISB	R0,-(R2)	;3RD INSTRUCTION WORD
	SWAB	R0
	BISB	R0,-(R2)	;INSERT LSB
	MOV	#INSTR,R1
	MOV	#XORW,R2	;INVERSION MASK
	MOV	#4,R3		;4 TIMES
12$:	MOV	(R2)+,R0	;GET MASK
	XOR	R0,R1
	DEC	R3		;DONE?
	BNE	12$		;NO

	MOV	(SP)+,R5
	RTS	PC
�
UNBLD::
	CLRB	PSUDOF		;RESET PSUEDO FLAG
	CLRB	BCONT		;BRANCH CONTROL
	CLRB	FUNC		;ALU FUNCTION
	CLRB	SORC		;ALU SOURCE
	CLRB	DEST		;ALU DESTINATION
	CLRB	AADDR		;A ADDRESS
	CLRB	BADDR		;B ADDRESS
	CLR	IDTA		;IMMEDIATE DATA
	CLRB	SCONT		;SEQUENCE CONTROL
	CLRB	CCONT		;CARRY CONTROL
	CLRB	SHIFT		;SHIFT CONTROL
	CLRB	IMC		;IMMEDIATE DATA CONTROL
	CLRB	PFUNC		;FUNCTION FLAG
	CLRB	PSCONT		;SEQUENCE CONTROL FLAG
UNBLD2::
	MOV	#INSTR,R0
	CLR	(R0)+		;CLEAR INSTRUCTION STORAGE
	CLR	(R0)+
	CLR	(R0)+
	CLR	(R0)
	RTS	PC
�
;  TRANSFER INSTRUCTION TO HEX. ASCII
BLST::
	TSTB	COND		;CODE ON?
	BNE	3$		;NO

	TSTB	PSUDOF		;PSUEDO OP?
	BMI	3$		;YES, COMMENT

	BNE	2$		;YES
	MOV	#LOC,R3		;CURRENT PC
	MOV	#CLINE+4,R4	;LISTING
	MOV	#4,R5		;# BYTES
	JSR	PC,BINHX	;CONVERT TO HEX. ASCII

	MOV	#INSTR,R3	;INSTRUCION WORD
1$:	MOV	#CLINE+9.,R4	;LISTING
	MOVB	.BCN,R5		;WORD BIT COUNT
	ASR	R5		;/2
	ASR	R5		;/4 = CHAR. COUNT
	JSR	PC,BINHX	;CONVERT TO HEX ASCII

3$:	JSR	PC,LINPG	;CHECK END OF PAGE
	RTS	PC

2$:	MOV	#PSUDOP,R3	;PRINT PSUEDO VALUE
	BR	1$


;  NOW OUTPUT THE OBJECT CODE
PUTBIN::
	TSTB OBJF     ;OBJECT OUTPUT FILE?
	BNE	2$		;YES

	RTS	PC		;NO

2$:	TSTB	GSDHD		;GSD HEADER BLOCK?
	BNE	7$		;NO

	MOVB	.BCN,R0		;INST. BIT COUNT
	ASR	R0		;/2
	ASR	R0		;/4
	ASR	R0		;/8=BYTE COUNT
	MOV	R0,R1		;SAVE BYTE COUNT
	MOV	R0,BLEN		;SET BLOCK LENGTH
	ADD	#INSTR,R0
	MOV	R0,IEND		;SET INSTRUCION END
	MOV	MLOC,R2		;MAX. LOCATION
	SUB	LOC,R2		;MIN. LOCATION
	INC	R2		;ONE MORE JUST CAUSE
	CLR	R0		;INIT TOTAL BYTE COUNT
3$:	ADD	R1,R0		;HARD WAY MULTIPLY
	DEC	R2
	BNE	3$
	ADD	#8.,R0		;8 MORE FOR HEADER
	MOV	R0,GSDWD	;SET WORD COUNT

	MOV	#GSDBLK,R0
	CLR	R2
4$:	MOVB	(R0)+,R1	;BUILD CHECKSUM BYTE.
	BIC	#177400,R1
	ADD	R1,R2
	CMP	R0,#GCK		;END OF BLOCK?
	BLO	4$		;NO

	MOVB	R2,GCK		;SET GSD CHECKSUM.
	NEGB	GCK		;MAKE IT NEG.
	MOV	BBPTR,R4	;CURRENT BUF ADRS.
	MOV	#GSDBLK,R0	;GSD HEADER BLOCK ADRS.
�
5$:	MOVB	(R0)+,(R4)+	;PUT GSD HEADER DIRECTLY IN BUF.
	CMP	R0,#EGSDBK	;END OF BLOCK?
	BLO	5$		;NO

	MOV	R4,BBPTR	;SET TO CURRENT BUF ADRS.
	INCB	GSDHD		;SET GSD HEADER DONE.
	JSR	PC,TXTHD	;PUT TXT HEADER IN BIN BUF.

	MOV	#LOC,R0		;PUT BEGIN/END ADRS IN BUF.
6$:	MOVB	(R0),R1
	MOVB	CKSUM,R2
	ADD	R1,R2
	MOVB	R2,CKSUM
	MOVB	(R0)+,(R4)+
	CMP	R0,#LOC+8.	;END OF HEADER?
	BLO	6$		;NO

	MOV	R4,BINPTR

7$:	MOV	#INSTR,R0	;NEXT INSTR.
	MOV	BINPTR,R4	;CURRENT BIN BLOCK PTR.
8$:	CMP	R4,#EBINBK	;MAX BYTES IN BLOCK?
	BLO	10$		;NO

	MOV	R0,-(SP)	;SAVE R0.
	SUB	#BINBLK,R4	;CALCULATE TOTAL BYTE COUNT.
	MOVB	CKSUM,R0
	ADD	R4,R0		;ADD TO CKSUM
	MOVB	R0,CKSUM
	MOVB	R4,@#BINBLK+2	;SET BYTE COUNT IN BUF.
	SUB	#8.,R4
	ADD	R4,LDARS	;ADD TO LOAD ADRS FOR NEXT TXT BLK.
	NEGB	CKSUM		;SET CKSUM NEG.
	MOV	BBPTR,R4	;PUT BIN BLOCK INTO OUTPUT BUF.
	MOV	#BINBLK,R0
9$:	JSR	PC,PUTB		;TEST BUF FULL.

	MOVB	(R0)+,(R4)+
	CMP	R0,#CKSUM+1	;END OF BINARY BLOCK?
	BLO	9$		;NO

	MOV	R4,BBPTR	;SAVE CURRENT BUF ADRS.

	JSR	PC,TXTHD	;PUT TXT HEADER IN BIN BUF.

	MOV	(SP)+,R0	;RESTORE R0.
10$:	MOVB	(R0),R1		;ADD BYTE TO CKSUM
	MOVB	CKSUM,R2
	ADD	R1,R2
	MOVB	R2,CKSUM
	MOVB	(R0)+,(R4)+	;PUT IN BINARY BLOCK BUF.
	CMP	R0,IEND		;END OF INSTRUCTION?
	BLO	8$		;NO.

	MOV	R4,BINPTR	;SAVE CURRENT BIN PTR.
	RTS	PC		;EXIT
�
PUTB:	CMP	R4,#EBINBF	;BUF FULL?
	BLO	2$		;NO.

	MOV	R0,-(SP)
	MOV	#BPAR,R0
	MOV	R5,-(SP)
	.WRITW			;WRITE THE BUFFER
	MOV	(SP)+,R5
	BCS	3$		;ERROR

	INC	BBLK		;NEXT BLOCK
1$:	MOV	(SP)+,R0
	MOV	#BINBUF,R4	;INIT BUF ADRS.
2$:	RTS	PC		;EXIT.

3$:	JSR	PC,POERR	;PRINT ERROR

	CLRB	OBJF		;TURN OFF OBJECT
	BR	1$

POERR::
	TST	R0		;EOF ERROR?
	BNE	4$		;NO

	.PRINT	#5$		;EOF ERROR
	RTS	PC

4$:	.PRINT	#6$		;OTHER ERROR
	RTS	PC

5$:	.ASCII	<15><12>'ASM - OUTPUT FULL ERROR'<200>
6$:	.ASCII	<15><12>'ASM - OUTPUT ERROR'<200>
	.EVEN



; INSERT TXT HEADER INTO BINARY
; BLOCK BUFFER.
TXTHD::
	MOV	#TXTBLK,R0	;TXT HEADER ADRS.
	MOV	#BINBLK,R4	;BINARY BLOCK ADRS.
	CLRB	CKSUM		;CLEAR CHECKSUM.
1$:	MOVB	(R0),R2		;ADD BYTE TO CHECKSUM.
	MOVB	CKSUM,R3
	ADD	R2,R3
	MOVB	R3,CKSUM
	MOVB	(R0)+,(R4)+	;PUT TXT HEADER INTO BUF.
	CMP	R0,#ETXTBK	;END OF TXT BLOCK HEADER?F
	BLO	1$		;NO.

	RTS	PC		;EXIT
�
;  CALL "SRC" BEFORE CALLING "PSUDO"
;	JSR	R5,SRC		;SEARCH FOR PSUEDO OP
;	.WORD	PSRC,NOTFND	;PSUEDO OP TABLE, NOT FOUND ADDR.
;
;  R1 = PSUEDO OP CODE NUMBER
;	JSR	PC,PSUDO	;PROCESS THE PSUEDO OP

PSUDO::
	INCB	PSUDOF		;PSUEDO OP FLAG
	MOVB	R1,PSUDOV	;PSUEDO OP CODE VALUE
	MOV	#PSUDOP,R0
	CLR	(R0)+		;CLEAR INSTRUCTION
	CLR	(R0)+
	CLR	(R0)+
	CLR	(R0)
	ASL	R1		;CHANGE OP CODE TO WORD OFFSET
	TSTB	COND		;CODE ON?
	BEQ	2$		;YES

	CMP	R1,#4*2		;CODE OFF, "IF"?
	BHI	3$		;NOT AN "IF"

2$:	JMP	@1$(R1)		;GO TO PSUEDO OP ROUTINE

1$:	.WORD	PSIF	;0 IF CONDITIONAL
	.WORD	PSENC	;1 END OF CONDITIONAL
	.WORD	PSIFF	;2 IF CONDITIONAL FALSE
	.WORD	PSIFT	;3 IF CONDITIONAL TRUE
	.WORD	PSIFTF	;4 IF CONDITIONAL TRUE OR FALSE
	.WORD	PSEQ	;5 EQUATE
	.WORD	PSTTL	;6 TITLE STATEMENT
	.WORD	PSGL	;7 GLOBLE PROGRAM NAME
	.WORD	PSPG	;8 PAGE LINE COUNT
	.WORD	PSEN	;9 PROGRAM END
	.WORD	PSRG	;10 PROGRAM ORIGIN
	.WORD	PSLST	;11 LIST STATEMENT
	.WORD	PSNLS	;12 NOLIST STATEMENT
	.WORD	PDEF	;13 DEFINE INSTR. FIELDS
	.WORD	PSLSB	;14 START OF LOCAL SYMBOL BLOCK
	.WORD	PELSB	;15 END OF LOCAL SYMBOL BLOCK
	.WORD	PXOR	;16 INSTR. POLARITY MASK WORD
	.WORD	PREP	;17 START REPEAT LOOP
	.WORD	PEREP	;18 END REPEAT LOOP
	.WORD	PIMSK	;19 IMMEDIATE DATA MASK

3$:
PSLSB::
PREP::
PEREP::
PELSB::
	MOVB	#-1,PSUDOF	;COMMENT FLAG
	RTS	PC		;DUMMY ROUTINES FOR NOW
�
;  SYMBOL EQUATE STATEMENT
PSEQ::
	TSTB	PLBLF		;WAS THERE A LABEL?
	BNE	2$		;YES

1$:	ERR	'L		;LABEL ERROR
	RTS	PC

2$:	TSTB	LTYP		;LOCAL LABEL?
	BEQ	1$		;YES = ERROR

;  "SLBL" JUST SET A NEW VALUE IN "JSTMD" BUT
;  "EQU" STATEMENTS DON'T DO LOCAL SECTION
;  UPDATES SO "JSTMD" MUST BE RESTORED.
	MOV	OLDMD,JSTMD	;RESTORE JSTMD
	DECB	PLBLF		;NEW LABEL?
	BEQ	63$		;NO
	JSR	R5,GARG		;GET THE EQUATE ARGUMENT
	.WORD	6$		;ERROR ADDRESS

	MOV	VADD,R3		;SYMBOL TABLE VALUE POINTER
	MOV	#PSUDOP,R4	;RDY FOR LISTING
	MOV	R1,(R3)+	;INSERT LSB 16 BITS
	MOV	R1,(R4)+
	BICB	#177,-3(R3)	;SAVE ERROR FLAG
	BISB	R0,-3(R3)	;CHANGE THE LABEL TYPE
	ASR	R0		;CHANGE TO WORD COUNT
	BR	5$

4$:	MOV	(R2),(R3)+	;SET VALUE IN ENTRY
	MOV	(R2)+,(R4)+	;SET IN LISTING
5$:	DEC	R0		;DONE?
	BNE	4$		;NO

	CLR	(R3)		;MARK THE END OF TABLE
	RTS	PC

6$:	MOV	VADD,R3		;VALUE ADDRESS
	CLR	-6(R3)		;CHANGE THE ENTRY TO END FLAG
	ERR	'U		;PASS 2 = UNDEFINED REF.
	RTS	PC

63$:	TSTB	PASS		;PASS1?
	BEQ	62$		;YES = MULT. DEF. ERROR

	MOVB	LTYP,R0		;NO, GET TYPE
	ASR	R0		;CHANGE TO WORD COUNT
	MOV	VADD,R3		;SYM. TBL. VALUE POINTER
	MOV	#PSUDOP,R4
61$:	MOV	(R3)+,(R4)+	;SAVE FOR LISTING
	DEC	R0
	BNE	61$

	RTS	PC

62$:	ERR	'M		;NO = MULTIPLE DEFINITION
	RTS	PC
�
;  TITLE STATEMENT
PSTTL::
	MOVB	#-1,PSUDOF	;LIST AS A COMMENT
	MOV	SPTR,R3		;SOURCE FIELD POINTER
	MOV	#HDR1,R2	;TITLE AREA
	MOV	#47.,R1		;MAX. BYTE COUNT
1$:	MOVB	(R3)+,R0	;GET A CHARACTER
	CMP	R0,#40		;SPACE?
	BEQ	1$		;YES, SKIP IT

	CMP	R0,#11		;TAB?
	BEQ	1$		;YES, SKIP IT

	TSTB	-(R3)		;BACK UP TO IT
	BR	3$		;COPY IT OVER

2$:	MOVB	(R3)+,(R2)+
3$:	TSTB	(R3)		;END OF LINE?

	BMI	4$		;YES
	DEC	R1		;END OF TITLE?

	BNE	2$		;NO
4$:	RTS	PC


PSEN::
	FLAG	ENDF		;SET END OF PROGRAM FLAG
	MOV	#STRT,-(SP)	;STARTING ADDRESS
	BR	PSET		;SET THE PSUEDO OP VALUE

PSET::
	JSR	R5,GARG		;GET THE ARGUMENT
	.WORD	PSETR		;ERROR
	MOV	R1,@(SP)+	;SET THE VALUE
PSET1::
	MOV	R1,PSUDOP	;SAVE FOR LISTING
	RTS	PC
PSETR::
	TST	(SP)+		;POP
PSERR	= .
	ERR	'B		;BAD ARGUMENT
	RTS	PC

PSPG::
	MOV	#LCNT,-(SP)	;READY TO SET LINE COUNT
	BR	PSET

PIMSK::
	JSR	R5,GARG		;GET ARGUMENT
	.WORD	PSETR		;ERROR
	MOV	R1,PSUDOP	;SAVE FOR LISTING
	COM	R1		;READY TO BE A MASK
	MOV	R1,IDTAM	;SAVE MASK
	RTS	PC
�
;	.DEF	.BRC.,29.	;SET ".BRC"=29.
PDEF::
	FIELD	1$,PDEF,2$,1$,1$
1$:	ERR	'B		;BAD ARGUMENT
	RTS	PC

2$:	JSR	R5,SRC		;FIND THE ".DEF" OPTION
	.WORD	DTBL,1$		;OPT. TBL., ERR ADDR
	MOV	R1,-(SP)	;SAVE SET OPTION VALUE
	JSR	R5,GARG		;GET VALUE IN R1

	.WORD	3$		;ERROR
	MOV	(SP)+,R0	;OPTION TABLE OFFSET
	MOVB	R1,FLDVL(R0)	;SET NEW OPTION VALUE
	MOV	R1,PSUDOP	;SAVE FOR LISTING
	RTS	PC

3$:	TST	(SP)+		;POP
	BR	1$


;	.ORG	XX		;SET LOC TO XX
PSRG::
	JSR	R5,GARG		;GET ARG. IN R1
	.WORD	PSERR		;ERROR

	TST	LOC		;START UP?
	BEQ	4$		;YES

	CMP	R1,LOC		;ORG BACKWARD?
	BHIS	1$		;NO

	ERR	'O		;ORG ERROR
	RTS	PC

1$:	MOV	R1,-(SP)
	JSR	PC,UNBLD	;ZERO THE INSTRUCTION

2$:	CMP	LOC,(SP)	;DONE?
	BHIS	3$		;YES

	TSTB	PASS		;PASS 1?
	BEQ	5$		;YES

	JSR	PC,PUTBIN	;OUTPUT ONE NOP

5$:	INC	LOC
	BR	2$

3$:	MOV	(SP)+,R1
4$:	MOV	R1,LOC		;SET ORGIN
	MOV	R1,PSUDOP
	FLAG	PSUDOF		;SET PSUEDO FLAG BACK ON
	RTS	PC

�
PSGL::
	FIELD	1$,PSGL,2$,1$,1$
1$:	ERR	'G		;GLOBL NAME ERROR
	RTS	PC

2$:	SUB	R1,R2		;BYTE COUNT
	CMP	R2,#6		;>6?
	BLOS	3$		;NO

	MOV	#6,R2		;YES, LIMIT TO 6
3$:	JSR	PC,R50WD	;CONVERT 1ST 3 TO RAD50

	MOV	R4,GLOB		;SAVE 1ST WORD
	JSR	PC,R50WD	;CONVER 2ND 3

	MOV	R4,GLOB+2	;SAVE 2ND WORD
	RTS	PC


;  GET INSTRUCTION POLARITY MASK
PXOR::
	JSR	R5,GARG		;GET ARGUMENT
	.WORD	2$		;ERROR

	TST	-(R2)		;MOVE TO START OF VALUE
	MOV	#XORW,R0	;PLACE TO SAVE IT
	MOV	#PSUDOP,R4	;SAVE FOR PRINTOUT
	MOV	#4,R3		;# WORDS
1$:	MOV	(R2),(R0)+	;SAVE IT
	MOV	(R2)+,(R4)+	;SAVE IT
	DEC	R3		;DONE?
	BNE	1$		;NO

	RTS	PC

2$:	ERR	'B		;BAD PARAMETER
	RTS	PC

;  IF CONDITIONAL
PSIF::
	FIELD	1$,PSIF,2$,1$,1$
1$:	RTS	PC

2$:	JSR	R5,SRC		;FIND "IF" TERM
	.WORD	IFTRM,1$	;TBL, NOT FOUND

	MOV	R1,-(SP)	;SAVE "IF" TERM CODE
	CLR	-(SP)		;SET FOR "DEFINED"
	JSR	R5,GARG		;GET THE TERM ARG.
	.WORD	4$		;ERROR OR NOT FOUND

3$:	SUB	#2,R0		;COUNT IT
	BLE	5$		;YES

	BIS	(R2)+,R1	;NO GET ANOTHER ONE
	BR	3$

4$:	INC	(SP)		;SET FOR "NOT DEFINED"
5$:	MOV	(SP)+,R3
	ASL	(SP)		;MAKE WORD OFFSET
	ADD	#6$,(SP)	;GET TERM INDEX
	MOV	@(SP)+,PC	;GOTO TERM PROCESSOR
6$:	.WORD	IFEQ		;EQ
	.WORD	IFNE		;NE
	.WORD	IFDF		;DF
	.WORD	IFNDF		;NDF
IFEQ::
	TST	R1		;EQ?
	BEQ	IFON		;EQ TRUE
IFOFF::
	MOVB	#-1,COND	;TURN CODE OFF
	BR	IFSV
IFON::
	CLRB	COND		;TURN CODE ON
IFSV::
	MOVB	COND,CONDS	;SAVE THE CONDITION
	RTS	PC
IFNE::
	TST	R1
	BEQ	IFOFF		;NE FALSE

	BR	IFON		;NE TRUE
IFDF::
	TST	R3
	BEQ	IFON		;DF TRUE

	BR	IFOFF		;DF FALSE
IFNDF::
	TST	R3
	BEQ	IFOFF		;NDF FALSE

	BR	IFON		;NDF TRUE

PSIFF::
	MOVB	CONDS,COND
	COMB	COND		;SAVED COND. FALSE
	RTS	PC

PSIFT::
	MOVB	CONDS,COND	;SAVED COND. TRUE
	RTS	PC

PSENC::
PSIFTF::
	CLRB	COND		;TURN CODE ON.
	RTS	PC

PSLST::
	CLRB	NLST		;LISTING ON
	RTS	PC

PSNLS::
	BISB	#200,NLST	;LISTING OFF
	RTS	PC
�
;  R0 = ASCII ERROR CODE
;	JSR	PC,SERR		;SET ERROR CODE IN LISTING
SERR::
	TSTB	PASS		;PASS1?
	BEQ	3$		;YES

	MOV	R1,-(SP)
	MOV	#ERRS,R1	;ERROR POINTER
	CMPB	(R1)+,#40	;SPACE?
	BEQ	1$		;YES, SET THE ERROR

	CMPB	(R1)+,#40	;2ND ONE A SPACE?
	BNE	2$		;NO, JUST COUNT THIS ERROR

1$:	MOVB	R0,-(R1)	;SET THE ERROR CODE
2$:	INC	ERCNT		;COUNT IT
	MOV	(SP)+,R1
3$:	RTS	PC


;  R4 = SHIFT COUNT
;  R5 = CONTROL FIELD VALUE
;	JSR	PC,INSERT	;PUT FIELD VALUE IN INSTRUCTION
INSERT::
	MOV	#RSLT+6,R3	;WORK SPACE
	CLR	(R3)		;ZERO THE WORK SPACE
	CLR	-(R3)
	CLR	-(R3)
	MOV	R5,-(R3)	;PUT FIELD VALUE IN LSB
	JSR	PC,SHIFTL	;SHIFT FIELD INTO POSITION

	MOV	#INSTR+6,R0
	BIS	(R3),(R0)	;INSERT FIELD INTO INSTRUCTION
	BIS	-(R3),-(R0)
	BIS	-(R3),-(R0)
	BIS	-(R3),-(R0)
	RTS	PC
�
;X 1234 123456789ABCDE LABEL OPERATION EQUATES COMMENTS
LIST::
	TSTB	PASS		;PASS 1?
	BEQ	52$		;YES

	TSTB	PSUDOF		;PSUEDO OP?
	BEQ	10$		;NO

10$:	JSR	PC,PRINT	;DO LISTING

	JSR	PC,CLRLST	;CLEAR THE LISTING

52$:	TSTB	NLST		;TURN LISTING OFF?
	BEQ	53$		;NO

	BICB	#200,LSTF	;YES
	RTS	PC

53$:	BISB	#200,LSTF	;TURN LISTING ON
	RTS	PC


PRINT::
	MOV	#CLINE,R0	;LISTING ADDRESS
PRNT::
	TSTB	LSTF		;LISTING ON?
	BPL	3$		;NO

	TSTB	LISF		;OUTPUT FILE?
	BNE	1$		;YES

	CMP	#CLINE,R0	;LISTING LINE?
	BNE	3$		;NO, SKIP IT

	CMPB	CLINE+2,#40	;ANY ERROR?
	BEQ	3$		;NO, SKIP IT

	.PRINT			;NO, GO TO TT:

	BR	2$

1$:	CMPB	@STLIN,#16	;^N?
	BNE	4$		;NO

	CMP	#CLINE,R0	;LISTING LINE?
	BNE	4$		;NO, DON'T DO ^N

	MOV	STLIN,R0	;YES, SET NEW START OF LINE
	MOVB	#12,-(R0)	;SET LF
	MOVB	#15,-(R0)	;SET CR

4$:	JSR	PC,PLINE	;LISTING OUTPUT

	BCS	5$		;ERROR

2$:	DECB	LCNTR		;COUNT THE LINE
3$:	RTS	PC

5$:	JSR	PC,POERR	;PRINT ERROR

	CLRB	LISF		;TURN OFF LISTING
	BR	2$


LINPG::
	TSTB	LSTF		;LISTING ON?
	BPL	1$		;NO

	TSTB	LCNTR		;END OF PAGE?
	BPL	1$		;NO

	INC	PAGE		;NEXT PAGE
	MOV	PAGE,R3		;GET PAGE NO.
	MOV	#HDR3,R4	;ASCII DESTINATION
	MOV	#3,R5		;3 CHAR.
	JSR	PC,BINDEC	;CONVERT TO DECIMAL

	MOV	#HDR,R0
	JSR	PC,PRNT		;PRINT TITLE

	MOVB	LCNT,LCNTR	;RESET LINE COUNT
	MOVB	#14,HDR0	;DO FF'S FROM NOW ON
1$:	RTS	PC

HDR:	.ASCII	<15>
HDR0:	.ASCII	<15>
HDR1:	.REPT	48.
	.BYTE	40
	.ENDR

HDR2:	.ASCII	'AED META ASM REV.'<REV.>', PAGE '
HDR3:	.ASCII	'NNN'<15><12><12><200>
	.EVEN


CLRLST::
	MOV	#40,R0		;SPACE
	MOV	#CLINE+2,R1	;LISTING LINE
	MOV	#22./2,R2
1$:	MOVB	R0,(R1)+
	MOVB	R0,(R1)+
	DEC	R2		;DONE?
	BNE	1$		;NO

	RTS	PC
�
;  PSUEDO OP CODES
PSRCH::	FLD	EQU,5
	FLD	.IF,0
	FLD	.ENDC,1
	FLD	.IFF,2
	FLD	.IFT,3
	FLD	.IFTF,4
	FLD	.TITLE,6
	FLD	.GLOBL,7
	FLD	.PAGE,8.	;LINE COUNT PER PAGE
	FLD	.END,9.
	FLD	.ORG,10.
	FLD	.LIST,11.
	FLD	.NLIST,12.
	FLD	.DEF,13.
	FLD	.SLSB,14.	;*
	FLD	.ELSB,15.	;*
	FLD	.XOR,16.	;INSTR. POLARITY MASK
	FLD	.REPT,17.	;*  START REPEAT LOOP
	FLD	.ENDR,18.	;*  END OF REPEAT LOOP
	FLD	.IMASK,19.	;IMMEDIATE DATA MASK
;  *  NOT IMPLIMENTED YET.
	.BYTE	0
	.EVEN

�
;  DEFINITION TABLE
DTBL::	FLD	.BCN.,0		;BIT COUNT PER WORD
	FLD	.FUN.,1		;*
	FLD	.SRC.,2		;*
	FLD	.DES.,3		;*
	FLD	.AAD.,4		;*
	FLD	.BAD.,5		;*
	FLD	.BRC.,6		;BRANCH CONTROL FIELD
	FLD	.IDF.,7		;*
	FLD	.SQC.,8.	;*  SEQUENCE CONTROL FIELD
	FLD	.CCT.,9.	;*  CARRY CONTROL FIELD
	FLD	.SHF.,10.	;*
	FLD	.CLC.,11.	;CLEAR CARRY CONTROL
	FLD	.SEC.,12.	;SET CARRY CONTROL
	FLD	.ICF.,13.	;IMMEDIATE CONTROL FIELD
	FLD	.IMC.,14.	;IMMEDIATE CNTROL CODE
	FLD	.SEQ.,15.	;SEQUENCE CONTROL "NOP"
;  * NOT IMPLIMENTED YET.
	.BYTE	0
	.EVEN
�
BPAR:	.BYTE	0,11		;CHAN, WRITE CODE
BBLK:	.WORD	0,BINBUF,256.,0
GSDBLK:	.WORD	1,56,1		;GSD HEADER BLOCK
PGNM:	.WORD	101710,26710,0,0 ;DEFAULT PGM NAME= UCBUF.
	.WORD	127401,7624,410,0 ; . ABS.
	.WORD	0,0,450
GSDWD:	.WORD	0		;TOTAL WORD COUNT
GLOB:	.WORD	101672,102060,2150,0 ;DEFAULT GLOBL NAME =UCBUF.
	.WORD	127401,7624,1410,1
GCK:	.BYTE	0		;CKSUM GSD.
	.BYTE	1		;END GSD BLK.
	.WORD	3000,1000,173400
	.WORD	1,16,4,7,0,0,0	;RLD BLK.
	.BYTE	346
EGSDBK=.
	.EVEN
TXTBLK:	.WORD	1,0,3		;TXT BLOCK.
LDARS:	.WORD	0		;LOAD ADRS.
ETXTBK=.
BINPTR:	.WORD	0		;BINARY BLOCK PTR.
GSDHD:	.WORD	0		;GSD HEADER FLAG.
BINBLK:	.BLKB	46.		;BINARY BLOCK
EBINBK=.
CKSUM:	.BYTE	0
ENDBLK:	.BYTE	1,0,6,0,6,0,363,0
EENDBK=.
	.EVEN
BBPTR:	.WORD	BINBUF
	.EVEN
�

PCNTF:	.BYTE	0		;CONTINUATION FLAG
PLBLF:	.BYTE	0		;LABEL FLAG
PERRF:	.BYTE	0		;ERROR FLAG
POBJF:	.BYTE	0		;OBJECT FLAG
PEQUF:	.BYTE	0		;EQUATE FIELD FLAG
POUTF:	.BYTE	0		;OBJECT OUT FLAG
PNOP:	.BYTE	0		;NOP FLAG
PEQUS:	.BYTE	0		;= FLAG
PDEST:	.BYTE	0		;DESTINATION FLAG
PLFLG:	.BYTE	0		;LINE FLAG
PJADDR:	.BYTE	0
PCOND:	.BYTE	0
PFNOP:	.BYTE	0		;FUNCTION NOP
PCOMA:	.BYTE	0		;COMMA FLAG
PAADDR:	.BYTE	0
PBADDR:	.BYTE	0
PNEGF:	.BYTE	0		;NEG. TERM FLAG
PRVAL:	.BYTE	0		;R VALUE FLAG
PSVAL:	.BYTE	0		;S VALUE FLAG
PRV:	.BYTE	0
PSV:	.BYTE	0
PFUNC:	.BYTE	0		;FUNCTION FLAG
PSRC:	.BYTE	0		;SOURCE FLAG
PMRF:	.BYTE	0		;-R FLAG
PDTA1:	.BYTE	0		;DATA FLAG
PDTA3:	.BYTE	0		;DATA FLAG
PSCONT:	.BYTE	0		;SEQ. CONTROL FLAG
PCRY:	.BYTE	0		;CARRY FLAG
PCRRY:	.BYTE	0		;CARRY FLAG
PSPEC:	.BYTE	0
PJCND:	.BYTE	0		;JUMP CONDITION FLAG
SOCTF:	.BYTE	0		;OCTAL FLAG
SDECF:	.BYTE	0		;DECIMAL FLAG
SHEXF:	.BYTE	0		;HEX FLAG
SERRF:	.BYTE	0		;NUMBER ERROR FLAG
LTYP:	.BYTE	0		;VALUE TYPE FLAG
ENDF:	.BYTE	0		;END OF PROGRAM FLAG
PSUDOF:	.BYTE	0	;PSUEDO OP FLAG
PSUDOV:	.BYTE	0	;PSUEDO OP CODE
RV:	.BYTE	0
SV:	.BYTE	0
RVAL:	.BYTE	0
SVAL:	.BYTE	0
BCONT:	.BYTE	0	;BRANCH CONTROL
FUNC:	.BYTE	0	;ALU FUNCTION
SORC:	.BYTE	0	;ALU SOURCE
DEST:	.BYTE	0	;ALU DESTINATION
AADDR:	.BYTE	0	;A ADDRESS
BADDR:	.BYTE	0	;B ADDRESS
SCONT:	.BYTE	0	;SEQUENCE CONTROL
CCONT:	.BYTE	0	;CARRY CONTROL
SHIFT:	.BYTE	0	;SHIFT CONTROL
IMC:	.BYTE	0	;IMMEDIATE DATA CONTROL
PASS:	.BYTE	0	;PASS FLAG
OBJF:	.BYTE	0	;OBJECT FLAG
LISF:	.BYTE	0	;LISTING FLAG
FILE:	.BYTE	0	;CURRENT INPUT FILE NO.
EOFCNT:	.BYTE	0	;INPUT EOF FLAG
LCNTR:	.BYTE	0	;LINE COUNTER
COND:	.BYTE	0	;CODE ON/OFF FLAG
CONDS:	.BYTE	0	;SAVED COND.
NLST:	.BYTE	0	;NOLIST FLAG
	.EVEN
PAGE:	.WORD	0	;PAGE COUNTER
IDTA:	.WORD	0	;IMMEDIATE DATA
IDTAM:	.WORD	176000	;IMMEDIATE DATA MASK
VADD:	.WORD	0
ERCNT:	.WORD	0	;ERROR COUNT
IEND:	.WORD	0
SPTR:	.WORD	0	;"SCAN" LINE POINTER
SSYM:	.BLKW	2
RSLT:	.BLKW	4
PSUDOP:	.BLKW	4	;PSUEDO OP VALUE
LCNT:	.WORD	55.		;PAGE LINE COUNT
SYME:	.WORD	0	;SYMBOL TABLE END
SYMS:	.WORD	0	;SYMBOL TABLE START
SYML:	.WORD	0	;START OF LOCAL SYMBOL TBL
SYMLS:	.WORD	0	;LOCAL SECTION POINTER
JSTMD:	.WORD	0	;LOCAL SECTION BOUNDRY ADDRESS
OLDMD:	.WORD	0	;TEMP BACKUP FOR "JSTMD"
RPTR:	.WORD	.-.	;READ BUFFER POINTER
RPAR:	.BYTE	3,10		;CHANNEL, READ CODE
RBLK:	.WORD	0		;READ BLOCK #
	.WORD	RBUFF,256.,0	;BUFFER POINTER, AND SIZE
STLIN:	.WORD	0	;START OF LISTING TEXT

; 1ST 4 DATA WORDS
LOC:	.BLKW	4	;LOC = MIN. PC
MLOC	= LOC+2		;MAX. PC
STRT	= LOC+4		;EXECUTION START ADDRESS
BLEN	= LOC+6		;INSTRUCTION BIT LENGTH

XORW:	.WORD	0,0,0,0	;POLARITY MASK
INSTR:	.BLKW	4	;CURRENT INSTRUCION WORD
CLINE:	.ASCII	<15><12>'EE1234 123456789ABCDE '
CLINE2	= CLINE+9.
	.BLKW	40.
	.WORD	0
ERRS:	.ASCII	'  '		;ERROR CODE STORAGE
LINE	= ERRS+2
	.BLKW	40.
	.BYTE	0
	.EVEN
RBUFF:	.BLKW	256.
LSTBUF:	.BLKW	256.		;LISTING BUFFER
BINBUF:	.BLKW	256.		;OBJECT BUFFER
EBINBF:				;END OF OBJ BUFFER

	.BLKW	100.		;PATCH SPACE
	.BLKW	16.		;ROOM FOR ODT STACK
�
;	ASM ERROR CODES

;A DESTINATION MODIFIER ERROR
;B BAD ARGUMENT
;C CARRY FIELD ERROR
;D DESTINATION ERROR
;E EQUATE ERROR
;F FUNCTION ERROR
;G GLOBL ERROR
;H
;I
;J JUMP ADDRESS ERROR
;K JUMP CONDITION ERROR
;L LABEL ERROR
;M MULTIPLE DEFINITION
;N NUMBER ERROR
;O ORG ERROR
;P
;Q
;R FUNCTION "R" ERROR
;S FUNCTION "S" ERROR
;T TERM ERROR
;U UNDEFINED REFERENCE
;V
;W EXPRESSION ERROR
;X ASSEMBLER BUG
;Y
;Z



	.END	ASM

1280_Microcode/dpucod.asm
;
; 1280 Drawing Processor uCode
; A Kossow
; Feb 85
;
 .TITLE	Drawing Processor uCode Rev 1.0
 .DEF	.BCN.,40.
 .XOR	(RD+QREG+DPIDLE)
;
ZERO	EQU	0
;
; ALU A Path Reg Adrs
; Same as B Path Adrs, but shifted
; four more bits to the left
; Use:
;  R0 in A Data Path = (R0^A)
;
A	EQU	4
;
; ALU B Path Reg Adrs
;
R0 	EQU	ZERO
R1	EQU	(01^32.)
R2	EQU	(02^32.)
R3	EQU	(03^32.)
R4	EQU	(04^32.)
R5	EQU	(05^32.)
R6	EQU	(06^32.)
R7	EQU	(07^32.)
R8	EQU	(08^32.)
R9	EQU	(09^32.)
R10	EQU	(0A^32.)
R11	EQU	(0B^32.)
R12	EQU	(0C^32.)
R13	EQU	(0D^32.)
R14	EQU	(0E^32.)
R15	EQU	(0F^32.)
;
; Special names for the ALU registers
;
TMP	EQU	R0	; scratch reg
COLOR	EQU	R3	; color to write with
;
; ALU register names during vector draws..
;
ACC	EQU	R4	; Vect Accum
INC2	EQU	R5	; Breshenham incr 2
X	EQU	R6	; X (or dX) inputs
Y	EQU	R7	; Y (or dY) inputs
ABSX	EQU	R8	; abs val of x
PCNT	EQU	R8	; pixel counter
ABSY	EQU	R9	; abs val of y
;
; ALU register names during rectangle fills
;

;
; Clipping window boundaries
;
XWBOT	EQU	R10	; X window bottom
YWBOT	EQU	R11	; Y window bottom
XWTOP	EQU	R12	; X window top
YWTOP	EQU	R13	; Y window top
;
; Virtual -> Physical CAP offsets
;
XOFS	EQU	R14	; X offset
YOFS	EQU	R15	; Y offset
;
; Register Rd or Wrt bit
; Note: Field inverted by .XOR
;
WR	EQU	 ZERO
RD	EQU	(40^25.)
;
; Parameter Registers
;
PR0	EQU	 ZERO
PR1	EQU	(01^25.)
PR2	EQU	(02^25.)
PR3	EQU	(03^25.)
PR4	EQU	(04^25.)
PR5	EQU	(05^25.)
PR6	EQU	(06^25.)
PR7	EQU	(07^25.)
PR8	EQU	(08^25.)
PR9	EQU	(09^25.)
PRA	EQU	(0A^25.)
PRB	EQU	(0B^25.)
PRC	EQU	(0C^25.)
PRD	EQU	(0D^25.)
PRE	EQU	(0E^25.)
PRF	EQU	(0F^25.)
;
; Refresh Buffer Registers
;
XCAP	EQU	(10^25.)	; Physical X Cap
YCAP	EQU	(11^25.)	; Physical Y Cap
;
; Start of 2901-specific register addresses.
; These locations are not decoded during direct
; register accesses by the host, and are used for
; data paths availible only to the 2901.
;
CONST	EQU	(12^25.)	; 12 Bit Constant
DIO	EQU	(13^25.)	; Data to/from host
;	EQU	(14^25.)
;	EQU	(15^25.)
;	EQU	(16^25.)
;	EQU	(17^25.)
;	EQU	(18^25.)
;	EQU	(19^25.)
;	EQU	(1A^25.)
;	EQU	(1B^25.)
;	EQU	(1C^25.)
;	EQU	(1D^25.)
;	EQU	(1E^25.)
;	EQU	(1F^25.)
;
; End of 2901-specific registers
;
RDMSK	EQU	(20^25.)	; Read Mask
OVCLR	EQU	(21^25.)	; Overlay Color Register
LUADR	EQU	(22^25.)	; LUT Adr Register
LURED	EQU	(23^25.)	; LUT Red Data Register
LUGRN	EQU	(24^25.)	; LUT Grn Data Register
LUBLU	EQU	(25^25.)	; LUT Blu Data Register
WRMSK	EQU	(26^25.)	; Write Mask
CTL	EQU	(27^25.)	; Control Register
;
; Control Register Function Bits
;
MINDN	EQU	1		; Decr Minor Axis
MINUP	EQU	ZERO		; Incr Minor Axis
MAJDN	EQU	2		; Decr Major Axis
MAJUP	EQU	ZERO		; Incr Major Axis
XMAJ	EQU	ZERO		; X Is Major Axis
YMAJ	EQU	4		; Y Is Major Axis
V960	EQU	8		; Blank At line 959
DSPOVL	EQU	10		; Display 9th Plane
WRTOVL	EQU	20		; Write 9th Plane
BSYFLG	EQU	80		; Reset DONE to host
;
STPREG	EQU	(28^25.)	; Stipple Register
BR9	EQU	(29^25.)	; Byte reg 9
BR10	EQU	(2A^25.)	; Byte reg 10
BR11	EQU	(2B^25.)	; Byte reg 11
BR12	EQU	(2C^25.)	; Byte reg 12
BR13	EQU	(2D^25.)	; Byte reg 13
BR14	EQU	(2E^25.)	; Byte reg 14
BR15	EQU	(2F^25.)	; Byte reg 15
;
FLAGS	EQU	BR9		; DPU primitive options
;
; Flag byte bit definitions
;
DWC	EQU	80		; Double wide chr
DHC	EQU	40		; Double high chr
ESRC	EQU	20		; Erase src rect after copy
DMA	EQU	10		; Special DMA processing
;	EQU	8
CLP	EQU	4		; Enable clipping
MWE	EQU	2		; Enable 10 pixel wrts
STP	EQU	1		; Enable stippling
;
FGCLR	EQU	BR10		; Color to draw  with
BGCLR	EQU	BR11		; Color to erase with
;
; Registers specific to CHR instruction
;
CHRX	EQU	BR12		; Character width
CHRY	EQU	BR13		; Character height
CHRDX	EQU	BR14		; Inter-chr spacing (X)
;
; Registers specific to WCT instruction
;
CLTADR	EQU	BR12		; clr tbl entry to upd
CLTRED	EQU	BR13		; red  val to clr tbl
CLTGRN	EQU	BR14		; grn  val to clr tbl
CLTBLU	EQU	BR15		; blue val to clr tbl
;
; VRAM Access locations
;
VIDNOI	EQU	(30^25.)	; Access w/o adv caps
VIDMAJ	EQU	(31^25.)	; adv major axis
VIDMIN	EQU	(32^25.)	; adv minor axis
VIDBOTH	EQU	(33^25.)	; adv both
;
STPNOI	EQU	(34^25.)	; enable stippler
STPMAJ	EQU	(35^25.)
STPMIN	EQU	(36^25.)
STPBOTH	EQU	(37^25.)
;
MWENOI	EQU	(38^25.)	; enable 10 pixel wrts
MWEMAJ	EQU	(39^25.)
MWEMIN	EQU	(3A^25.)
MWEBOTH	EQU	(3B^25.)

;
; ALU Sources
;
AQ	EQU	ZERO
AB	EQU	(01^22.)
ZQ	EQU	(02^22.)
ZB	EQU	(03^22.)
ZA	EQU	(04^22.)
DA	EQU	(05^22.)
DQ	EQU	(06^22.)
DZ	EQU	(07^22.)
;
; ALU Operations
;
ADD	EQU	ZERO
SUBR	EQU	(01^16.)
SUBS	EQU	(02^16.)
OR	EQU	(03^16.)
AND	EQU	(04^16.)
NOTRS	EQU	(05^16.)
EOR	EQU	(06^16.)
EXNOR	EQU	(07^16.)
;
; ALU Carry Input Bit
;
SETC	EQU	(01^15.)
;
; ALU Destinations
; Note: LS Bit of field inv
; by .XOR
;
; Dest->     RAMB    Q    Y
;            ----    -    -
; QREG  (0)  -       F    F
; NOP   (1)  -       -    F
; RAMA  (2)  F       -    RAMA
; RAMF  (3)  F       -    F
; RAMQD (4)  F/2     Q/2  F
; RAMD  (5)  F/2          F
; RAMQU (6)  2F      2F   F
; RAMU  (7)  2F      -    F
;
QREG	EQU	(01^19.)
RAMF	EQU	(02^19.)
RAMA	EQU	(03^19.)
RAMD	EQU	(04^19.)
RAMQD	EQU	(05^19.)
RAMU	EQU	(06^19.)
RAMQU	EQU	(07^19.)
;
; Misc Control Bits
;
DPIDLE  EQU	(1^14.)	; DPU Busy bit
RMW	EQU	(1^13.)	; RM/W to vram sequencer
;
; Jump Conditions
;
JZ	EQU	(01^9.)	; jmp if zero
JNEG	EQU	(02^9.)	; jmp if negative
JCY	EQU	(03^9.)	; jmp if carry set
JLT	EQU	(04^9.)	; jmp if less than
;	EQU	(05^9.)
JNHBL	EQU	(06^9.)	; jmp if not h blank
JNVBL	EQU	(07^9.)	; jmp if not v blank
JNZ	EQU	(08^9.)	; jmp if not zero
JPOS	EQU	(09^9.)	; jmp if positive
JNCY	EQU	(0A^9.)	; jmp if carry clear
JGE	EQU	(0B^9.)	; jmp if gt or eq
;	EQU	(0C^9.)	
;	EQU	(0D^9.)	
;	EQU	(0E^9.)	
JMP	EQU	(0F^9.)	; jmp always

;
;	Macros for ALU operations
;
ANDAQ	EQU	(AND+AQ)	; F = A&Q
ANDAB	EQU	(AND+AB)	; F = A&B
ANDDA	EQU	(AND+DA)	; F = D&A
ANDDQ	EQU	(AND+DQ)	; F = D&Q
ORAQ	EQU	(OR+AQ)		; F = A|Q
ORAB	EQU	(OR+AB)		; F = A|B
ORDA	EQU	(OR+DA)		; F = D|A
ORDQ	EQU	(OR+DQ)		; F = D|Q
EORAQ	EQU	(EOR+AQ)	; F = A^Q
EORAB	EQU	(EOR+AB)	; F = A^B
EORDA	EQU	(EOR+DA)	; F = D^A
EORDQ	EQU	(EOR+DQ)	; F = D^Q
INVQ	EQU	(EXNOR+ZQ)	; F = !Q
INVB	EQU	(EXNOR+ZB)	; F = !B
INVA	EQU	(EXNOR+ZA)	; F = !A
INVD	EQU	(EXNOR+DZ)	; F = !D
ADDAQ	EQU	(ADD+AQ)	; F = A + Q
ADDAB	EQU	(ADD+AB)	; F = A + B
ADDDA	EQU	(ADD+DA)	; F = D + A
ADDDQ	EQU	(ADD+DQ)	; F = D + Q
SUBQA	EQU	(SETC+SUBR+AQ)	; F = Q - A
SUBBA	EQU	(SETC+SUBR+AB)	; F = B - A
SUBAD	EQU	(SETC+SUBR+DA)	; F = A - D
SUBQD	EQU	(SETC+SUBR+DQ)	; F = Q - D
SUBAQ	EQU	(SETC+SUBS+AQ)	; F = A - Q
SUBAB	EQU	(SETC+SUBS+AB)	; F = A - B
SUBDA	EQU	(SETC+SUBS+DA)	; F = D - A
SUBDQ	EQU	(SETC+SUBS+DQ)	; F = D - Q
INCQ	EQU	(SETC+ADD+ZQ)	; F = Q + 1
INCB	EQU	(SETC+ADD+ZB)	; F = B + 1
INCA	EQU	(SETC+ADD+ZA)	; F = A + 1
INCD	EQU	(SETC+ADD+DZ)	; F = D + 1
DECQ	EQU	(SUBR+ZQ)	; F = Q - 1
DECB	EQU	(SUBR+ZB)	; F = B - 1
DECA	EQU	(SUBR+ZA)	; F = A - 1
DECD	EQU	(SUBS+DZ)	; F = D - 1
NEGQ	EQU	(SETC+SUBS+ZQ)	; ~Q
NEGB	EQU	(SETC+SUBS+ZB)	; ~B
NEGA	EQU	(SETC+SUBS+ZA)	; ~A
NEGD	EQU	(SETC+SUBR+DZ)	; ~D
ALUZ	EQU	(AND+ZQ)	; F = 0
ALUD	EQU	(OR+DZ)		; F = D
ALUQ	EQU	(OR+ZQ)		; F = Q
ALUA	EQU	(OR+ZA)		; F = A
ALUB	EQU	(OR+ZB)		; F = B
ALUCONS EQU	(CONST+ALUD)	; F = CONSTANT

;+++++++++++++++++++++++++++++++++++++++++++++++++
; uCode jmp vectors:
; Unused entries should jmp to IDLE:
;+++++++++++++++++++++++++++++++++++++++++++++++++
; RST   - Reset the DPU
;         XCAP <- 0      YCAP <- 0
;         XBOT <- 0      YBOT <- 0
;         XTOP <- 1279   YTOP <- 1023
;         XOFS <- 0      YOFS <- 0
;         WRMSK<- ff     STP  <- ff
;         CTLRG<- 0
;-------------------------------------------------
	ALUZ			; aluopr = 0
+	XWBOT 			; aludst = XWBOT
+	WR XCAP			; aludst = XCAP
+	JMP RST
;-------------------------------------------------
; RWP   - Read Window Parameters
;         PR0 <- Bottom of window (X)
;         PR1 <- Bottom of window (Y)
;         PR2 <- Top    of window (X)
;         PR3 <- Top    of window (Y)
;         PR4 <- X Offset
;         PR5 <- Y Offset
;-------------------------------------------------
	XWBOT ALUB		; aluopr = XWBOT
+	WR PR0			; aludst = PR0
+	JMP RWP
;-------------------------------------------------
; WWP   - Write Window Parameters
;         Bottom of window (X) <- PR0
;         Bottom of window (Y) <- PR1
;         Top    of window (X) <- PR2
;         Top    of window (Y) <- PR3
;         X Offset             <- PR4
;         Y Offset             <- PR5
;-------------------------------------------------
	RD PR0 ALUD		; aluopr = PR0
+	RAMF XWBOT		; aludst = XWBOT
+	JMP WWP
;-------------------------------------------------
; DVA   - Draw Vector Absolute
;
;         FLAGS->clip = 4
;                mwe  = 2
;                stip = 1
;         XCAP-> start x
;         YCAP-> start y
;         PR0 -> end   x
;         PR1 -> end   y
;-------------------------------------------------
	RD PR0 ALUD		; aluopr = PR0
+	ALUD RAMF X		; aludst = X
+	JMP DVA 

;-------------------------------------------------
; DVR   - Draw Vector Relative
;
;         FLAGS->clip = 4
;                mwe  = 2
;                stip = 1
;         XCAP-> start x
;         YCAP-> start y
;         PR0 -> end x displ
;         PR1 -> end y displ
;-------------------------------------------------
	RD PR0 ALUD		; aluopr = PR0
+	RAMF X			; aludst = X
+	JMP DVR
;-------------------------------------------------
; DFR   - Draw Filled Rectangle
;
;         FLAGS->clip = 4
;                stip = 1
;         XCAP-> start x
;         YCAP-> start y
;         PR0 -> end x
;         PR1 -> end y
;         PR2 ->
;          .
;          .
;         PR9 -> Stipple pattern(s) 1-8
;-------------------------------------------------
	RD PR0 ALUD		; aluopr= PR0
+	RAMF X			; aludst = X
+	JMP DFR
;-------------------------------------------------
; FRR   - Filled Rectangle Relative
;
;         FLAGS->clip = 4
;                stip = 1
;         XCAP-> start x
;         YCAP-> start y
;         PR0 -> end x displ
;         PR1 -> end y displ
;         PR2 ->
;          .
;          .
;         PR9 -> Stipple pattern (1-8)
;-------------------------------------------------
	RD PR0 ALUD		; aluopr = PR0
+	RAMF X			; aludst = X
+	JMP FRR

;-------------------------------------------------
; CRA   - Copy Rectangle (absolute)
;
;         FLAGS->era src = 32.
;                clip    = 4
;
;         PR0 -> dstxtop (abs)
;         PR1 -> dstytop (abs)
;
;         PRC -> srcxbot (AOI...)
;         PRD -> srcybot
;         PRE -> srcxtop
;         PRF -> srcytop
;-------------------------------------------------
	RD PR0 ALUD
+	RAMF X
+	JMP CRA
;-------------------------------------------------
; CRR   - Copy Rectangle (relative)
;
;         FLAGS->era src = 32.
;                clip    = 4.
;
;         PR0 -> dstxtop (rel)
;         PR1 -> dstytop (rel)
;
;         PRC -> srcxbot (AOI...)
;         PRD -> srcybot
;         PRE -> srcxtop
;         PRF -> srcytop
;-------------------------------------------------
	RD PR0 ALUD
+	RAMF X
+	JMP CRR

;-------------------------------------------------
; CHR   - 12 * 12 special filled rect
;
;         FLAGS -> double width  = 128.
;                  double height = 64.
;         CHRX  -> number of pixels in x
;         CHRY  -> number of pixels in y
;         CHRDX -> x displacement to nxt chr
;         PR0
;          .
;          .
;         PR12 -> 12 bit stipple pattern(s)
;-------------------------------------------------
	JMP CHR
;-------------------------------------------------
;
; WPX - Write Pixel at current virtual position
;
;-------------------------------------------------
	JMP WPX
;-------------------------------------------------
;
; WMP - Write Multiple Pixels
;
;       FLAGS-> clip if flag != 0
;	XCAP -> starting x
;       YCAP -> starting y
;       PR0  -> number of pixels to wrt
;
;-------------------------------------------------
	RD CTL FLAGS		; aluopr = FLAGS
+	QREG			; aludst = Q
+	JMP WMP
;-------------------------------------------------
;
; RPX - Read Pixel at current virtual position
;
;-------------------------------------------------
	JMP RPX
;-------------------------------------------------
;
; WCT - Synchronized color table update
;       Update is performed at the begining
;       of the horz blanking interval.
;
;       CLTADR -> color table address
;       CLTRED -> red value
;       CLTGRN -> grn value
;       CLTBLU -> blu value
;
;       Returns CLTADR + 1
;-------------------------------------------------
	JMP WCT
;-------------------------------------------------
;
; ERW - Erase Window
;
;       BGCLR -> Color to erase with
;
;       XWBOT
;       YWBOT
;       XWTOP
;       YWTOP -> Rectangle to erase
;                (the currently defined window)
;
;       Note:
;             Wrt Mask is NOT changed
;-------------------------------------------------
;
; Set BG color as working color
;
	RD BGCLR ALUD		; aluopr = BGCLR
+	RAMF COLOR		; aludst = COLOR
+	JMP ERW
�
;-------------------------------------------------
;
; EFS - Edge Flag Search
;
;-------------------------------------------------
	JMP EFS
;-------------------------------------------------
;
; EFF - Edge Flag Fill
;
;-------------------------------------------------
	JMP EFF
;-------------------------------------------------
;
; IFS - Interior Fill Search
;
;-------------------------------------------------
	JMP IFS
;-------------------------------------------------
;
; IFF - Interior Fill Fill
;
;-------------------------------------------------
	JMP IFF
;-------------------------------------------------
;
; OFS - Overlay Fill Search
;
;-------------------------------------------------
	JMP OFS
;-------------------------------------------------
;
; OFF - Overlay Fill Fill
;
;-------------------------------------------------
	JMP OFF
;-------------------------------------------------
;
; GFS - General Fill Search
;
;-------------------------------------------------
	JMP GFS
;-------------------------------------------------
;
; GFF - General Fill Fill
;
;-------------------------------------------------
	JMP GFF

;
; DONE - Sets Done Flag to Host
;        Drops into RESCAP
;
DONE:
	RD CTL ALUD		; aluopr = CTL
+	QREG			; aludst = Q

	(ALUCONS+ANDDQ) #7F	; aluopr = Q & 7f
+	#QREG			; aludst = Q

	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
;
; RESCAP - Restore virtual CAP addr by subtracting
;          XOFS from XCAP and YOFS from YCAP
;          Drops into IDLE
;
RESCAP:
	#((RD+XCAP)+(XOFS^A)+SUBAD) ; aluopr = XCAP - XOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+XCAP)		    ; aludst = XCAP
	#((RD+YCAP)+(YOFS^A)+SUBAD) ; aluopr = YCAP - YOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+YCAP)		    ; aludst = YCAP
;
; DPU idle loop
; Control transfers here when there's
; nothing to do
;
; Note: Din read into the Q register
;
IDLE:
	#(RD+DIO+ALUD)		; aluopr = DIN
+	#QREG			; aludst = Q
+	#(JMP+IDLE) #DPIDLE
;
; RESET
;
RST:
	#ALUZ			; aluopr = 0
+	#(RAMF+YWBOT)		; aludst = YWBOT
+	#(WR+YCAP)		; aludst = YCAP
	#ALUCONS	#1279.	; aluopr = 1279.
+	#(RAMF+XWTOP)		; aludst = XWTOP
	#ALUCONS	#1023.	; aluopr = 1023.
+	#(RAMF+YWTOP)		; aludst = YWTOP
	#ALUZ			; aluopr = 0
+	#(RAMF+XOFS)		; aludst = XOFS
	#ALUZ			; aluopr = 0
+	#(RAMF+YOFS)		; aludst = YOFS
	#ALUZ			; aluopr = 0.
+	#(WR+CTL)		; aludst = CTL
	#ALUCONS	#0FF	; aluopr = ff
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+WRMSK)		; aludst = WRMSK
	#ALUQ			; aluopr = Q
+	#(WR+STP)		; aludst = STP
+	#(JMP+IDLE)		; back to sleep.
;
; RWP - Read Window Parameters
;
RWP:
	#(YWBOT+ALUB)		; aluopr = YWBOT
+	#(WR+PR1)		; aludst = PR1
	#(XWTOP+ALUB)		; aluopr = XWTOP
+	#(WR+PR2)		; aludst = PR2
	#(YWTOP+ALUB)		; aluopr = YWTOP
+	#(WR+PR3)		; aludst = PR3
	#(XOFS+ALUB)		; aluopr = XOFS
+	#(WR+PR4)		; aludst = PR4
	#(YOFS+ALUB)		; aluopr = YOFS
+	#(WR+PR5)		; aludst = PR5
+	#(JMP+IDLE)
;
; WWP - Write Window Parameters
;
WWP:
	#(RD+PR1+ALUD)		; aluopr = PR1
+	#(RAMF+YWBOT)		; aludst = YWBOT
	#(RD+PR2+ALUD)		; aluopr = PR2
+	#(RAMF+XWTOP)		; aludst = XWTOP
	#(RD+PR3+ALUD)		; aluopr = PR3
+	#(RAMF+YWTOP)		; aludst = YWTOP
	#(RD+PR4+ALUD)		; aluopr = PR4
+	#(RAMF+XOFS)		; aludst = XOFS
	#(RD+PR5+ALUD)		; aluopr = PR5
+	#(RAMF+YOFS)		; aludst = YOFS
+	#(JMP+IDLE)
;
; DVA - Draw Vector Absolute
; Convert X,Y to Dx,Dy and
; process with DVR
;
DVA:
	#(RD+PR1+ALUD)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
	#((X^A)+RD+XCAP+SUBAD)	; aluopr = X - XCAP
+	#(RAMF+X)		; aludst = X
	#((Y^A)+RD+YCAP+SUBAD)	; aluopr = Y - YCAP
+	#(RAMF+Y)		; aludst = Y
+	#(JMP+DVR1)
;
; DVR - Draw Vector Relative
;
DVR:
	#(RD+PR1+ALUD)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
;
; draw relative vector from x,y cap to r0,r1 (rel)
;
DVR1:
;
; Add X and Y displacements to CAPS and X,Y
;
	#((RD+XCAP)+(XOFS^A)+ADDDA) ; aluopr = XCAP + XOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+XCAP)		    ; aludst = XCAP
	#((RD+YCAP)+(YOFS^A)+ADDDA) ; aluopr = YCAP + YOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+YCAP)		    ; aludst = YCAP
	#((XOFS^A)+X+ADDAB)	    ; aluopr = XOFS + X
+	#(RAMF)			    ; aludst = X
	#((YOFS^A)+Y+ADDAB)	    ; aluopr = YOFS + Y
+	#(RAMF)			    ; aludst = Y
;
; Do Vector Initialization
;
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(ALUQ+ORDQ)	#BSYFLG	; aluopr = Q | BSYFLG
+	#QREG			; aludst = Q
	#(CONST+ANDDQ)	#0F8	; aluopr = Q & f8
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
	#(ABSX+ALUB)		; aluopr = ABSX
	#(ABSY+ALUB)		; aluopr = ABSY
+	#(JPOS+DVR2)		; jmp if ABSX pos

	#(ABSX+NEGB)		; aluopr = ABSX
+	#RAMF			; aludst = ~ABSX

	#(ABSY+ALUB)		; aluopr = ABSY
DVR2:
	#(JPOS+DVR3)		; jmp if ABSY pos

	#(ABSY+NEGB)		; aluopr = ABSY
+ 	#(RAMF)			; aludst = ~ABSY
DVR3:
	#((ABSY^A)+ABSX+SUBAB)	; aluopr = ABSY - ABSX
	#(X+ALUB)		; aluopr = X
+	#(JPOS+YMAJAXIS)	; jmp if ABSY-ABSX pos
	#(Y+ALUB)		; aluopr = Y
+	#(JNEG+XMAJDN)		; jmp if X negative
DVR4:
	#(ABSX+ALUB)		; aluopr = ABSX
+	#(RAMD+TMP)		; aludst = TMP
+	#(JNEG+YMINDN)		; jmp if Y negative
DVR5:
	#((TMP^A)+ABSY+SUBBA)	; aluopr = ABSY - TMP
+	#(RAMF+ACC)		; aludst = ACC
	#((ABSX^A)+ABSY+SUBAB)	; aluopr = ABSY - ABSX
+	#(RAMF+INC2)		; alout  = INC2
;
; Determine what type of vector to do
; norm, stippled, or mult-wrt
;
	#(RD+FLAGS+ALUD)	; aluopr = FLAGS
+	#QREG			; aludst = Q
	#(CONST+ANDDQ+STP+MWE)	; aluopr = Q &(STP|MWE)
	#(CONST+ANDDQ+STP)	; aluopr = Q&(STP)
+	#(JZ+NORMDVR)		; jmp if   Q&(STP|MWE) == 0
	#(JNZ+STIPDVR)		; jmp if   Q&(STP) != 0

;-----------------------------------------------
;
; Vector Draw Inner Loops
;
;-----------------------------------------------
;
; MWE Inner Loop
;
;-------------------->

	#(JMP+DONE)	; all done.. restore caps
;
; Stipple Inner Loop
;
;-------------------->
STIPDVR:
	#(JMP+DONE)	; all done.. restore caps
;
; Normal Inner Loop
NORMDVR:
;-------------------->
	#(JMP+DONE)	; all done.. restore caps
�
;
; Set Cap direction for y = major axis
;
YMAJAXIS:
;
; Set Y as major axis
;
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#YMAJ	; aluopr = Q | YMAJ
	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
;
; Check if X or Y counters must be set
; to decr
;
	#(Y+ALUB)		; aluopr = Y
	#(X+ALUB)		; aluopr = X
+	#(JNEG+YMAJDN)		; jmp if Y neg
YMAJ1:
	#(JNEG+XMINDN)		; jmp if X neg
;
; Swap ABSX and ABSY
;
YMAJ2:
	#(ABSX+ALUB)		; aluopr = ABSX
+	#QREG			; aludst = Q
	#((ABSY^A)+ALUA)	; aluopr = ABSY
+	#(RAMF+ABSX)		; aludst = ABSX
	#ALUQ			; aluopr = QREG
+	#(RAMF+ABSY)		; aludst = ABSY
	#(ABSX+ALUB)		; aluopr = ABSX/2
+	#(RAMD+TMP)		; aludst = TMP
+	#(JMP+DVR5)
;
; Set Y cap direction down (major axis)
;
YMAJDN:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#MAJDN  ; aluopr = Q | MAJDN
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
	#(X+ALUB)		; aluopr = X
+	#(JMP+YMAJ1)
;
; Set X cap dir down (minor axis)
;
XMINDN:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#MINDN	; aluopr = Q | MINDN
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
+	#(JMP+DVR5)
;
; Set X cap dir down (major axis)
;
XMAJDN:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#MAJDN	; aluopr = Q | MAJDN
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
	#(Y+ALUB)		; aluopr = Y
+	#(JMP+DVR4)
;
; Set Y cap dir down (minor axis)
;
YMINDN:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#MINDN	; aluopr = Q | MINDN
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
	#(ABSX+ALUB)		; aluopr = ABSX/2
+	#(RAMD+TMP)		; aludst = TMP
+	#(JMP+DVR5)
�
;
; DFR - Draw Filled Rectangle
;
; Convert Absoute X,Y to relative and
; process using FRR
;
DFR:
	#(RD+PR1+ALUD)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
	#((X^A)+RD+XCAP+SUBAD)	; aluopr = X - XCAP
+	#(RAMF+X)		; aludst = X
	#((Y^A)+RD+YCAP+SUBAD)	; aluopr = Y - YCAP
+	#(RAMF+Y)		; aludst = Y
+	#(JMP+FRR1)
;
; FRR - Filled Rectangle Relative
;
FRR:
	#(RD+PR1)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
;
; Draw filled rectangle from ... to DX,DY
;
FRR1:
;
; Get color to write in
;
	#(RD+FGCLR+ALUD)	; aluopr = FGCLR
+	#(RAMF+COLOR)		; aludst = COLOR
FRR2:
;
; Add X and Y displacements to CAPS and X,Y
;
	#((RD+XCAP)+(XOFS^A)+ADDDA) ; aluopr = XCAP + XOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+XCAP)		    ; aludst = XCAP
	#((RD+YCAP)+(YOFS^A)+ADDDA) ; aluopr = YCAP + YOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+YCAP)		    ; aludst = YCAP
	#((XOFS^A)+X+ADDAB)	    ; aluopr = XOFS + X
+	#(RAMF)			    ; aludst = X
	#((YOFS^A)+Y+ADDAB)	    ; aluopr = YOFS + Y
+	#(RAMF)			    ; aludst = Y
;
 	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#BSYFLG	; aluopr = Q | BSYFLG
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL

	#(JMP+DONE)		; reset CAPS to virt loc
�
;
; CRA - Copy Rectangle Absolute
;
; Convert Absoute X,Y to relative and
; process using CRR
;
CRA:
	#(RD+PR1+ALUD)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
	#((X^A)+RD+XCAP+SUBAD)	; aluopr = X - XCAP
+	#(RAMF+X)		; aludst = X
	#((Y^A)+RD+YCAP+SUBAD)	; aluopr = Y - YCAP
+	#(RAMF+Y)		; aludst = Y
+	#(JMP+FRR1)
;
; CRR - Copy Rectangle Relative
;
CRR:
	#(RD+PR1)		; aluopr = PR1
+	#(RAMF+Y)		; aludst = Y
;
; Add X and Y displacements to CAPS and X,Y
;
CRR1:
	#((RD+XCAP)+(XOFS^A)+ADDDA) ; aluopr = XCAP + XOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+XCAP)		    ; aludst = XCAP
	#((RD+YCAP)+(YOFS^A)+ADDDA) ; aluopr = YCAP + YOFS
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+YCAP)		    ; aludst = YCAP
	#((XOFS^A)+X+ADDAB)	    ; aluopr = XOFS + X
+	#(RAMF)			    ; aludst = X
	#((YOFS^A)+Y+ADDAB)	    ; aluopr = YOFS + Y
+	#(RAMF)			    ; aludst = Y
;
; Go busy..
;
 	#(RD+CTL+ALUD)		    ; aluopr = CTL
+	#QREG			    ; aludst = Q
	#(CONST+ORDQ)	#BSYFLG	    ; aluopr = Q | BSYFLG
+	#QREG			    ; aludst = Q
	#ALUQ			    ; aluopr = Q
+	#(WR+CTL)		    ; aludst = CTL
�
;
; Check if rectangles overlap, and determine correct
; copy order
;
;       +---+               +---+                +---+
; dx<0  |DDD|          dx=0 |DDD|        dx>0    |DDD|
; dy>0  |DDD|-+        dy>0 |DDD|        dy>0  +-|DDD|
;       +---+S|             +---+              |S+---+
;         |SSS|             |SSS|              |SSS|
;         +---+             +---+              +---+
;
;       +---+-+             +---+              +-+---+
; dx<0  |DDD|S|        dx=0 |DDD|        dx>0  |S|DDD|
; dy=0  |DDD|S|        dy=0 |DDD|        dy=0  |S|DDD|
;       +---+-+             +---+              +-+---+
;
;         +---+             +---+              +---+
; dx<0    |SSS|        dx=0 |SSS|        dx>0  |SSS|
; dy<0  +---+S|        dy<0 +---+        dy<0  |S+---+
;       |DDD|-+             |DDD|              +-|DDD|
;       |DDD|               |DDD|                |DDD|
;       +---+               +---+                +---+
;

;
; Loop for rectangle copy
; w/o src rectangle erase
;

;
; Loop for rectangle copy
; with src rectangle erase
;

	#(JMP+DONE)		; reset CAPS to virt loc

�
;
; CHR - Draw Character
;
CHR:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#BSYFLG	; aluopr = Q | BSYFLG
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL

	#(JMP+DONE)
�
;
; WPX - Write Pixel
;
WPX:
;
; Check if this is the last pixel in an
; AOI operation. Set DONE if it was.
;
CHKAOI:
;
; Wasn't last pixel... Don't set DONE
;
	#(JMP+RESCAP)
;
; Was the last pixel.. Set DONE
;
	#(JMP+DONE)
;
; WMP - Write Multiple Pixels
;
WMP:
	#(RD+PR0+ALUD)	; aluopr = PR0
+	#(RAMF+PCNT)	; aludst = PCNT
+	#(JZ+WMP1)	; flag byte not set
;
; clip pixel run to window
;

;
; Set BUSY
;
WMP1:
	#(RD+CTL+ALUD)		; aluopr = CTL
+	#QREG			; aludst = Q
	#(CONST+ORDQ)	#BSYFLG	; aluopr = Q | BSYFLG
+	#QREG			; aludst = Q
	#ALUQ			; aluopr = Q
+	#(WR+CTL)		; aludst = CTL
;
; start of write loop
;
WMPLP:	#(PCNT+DECB)		; aluopr = PCNT-1
+	#((COLOR^A)+RAMA)	; aludst = PCNT
+	#(WR+VIDMAJ)		; COLOR -> Vram
	#(JNZ+WMPLP)		; tst done in 2nd half
	#(JMP+DONE)		; pop out when done
;
; RPX - Read Pixel
;
RPX:

;
; Check to see if this is the end of AOI
; operation
;
	#(JMP+CHKAOI)
�
;
; WCT - Synchronized Color Table Access
;
WCT:
	#(JMP+DONE)
;
; ERW - Erase Window
;
ERW:
;
; x and y rectangle size is the size of the window
;
	#ALUZ			 ; aluopr = 0
+	#(WR+XCAP)		 ; aludst = XCAP
	#ALUZ			 ; aluopr = 0
+	#(WR+YCAP)		 ; aludst = YCAP
	#((XWTOP^A)+XWBOT+SUBAB) ; aluopr = XWTOP-XWBOT
+	#(RAMF+X)		 ; aludst = X (dx)
	#((YWTOP^A)+YWBOT+SUBAB) ; aluopr = YWTOP-YWBOT
+	#(RAMF+Y)		 ; aludst = Y (dy)
+	#(JMP+FRR2)		 ; 'draw' rectangle
�
;
; EFS - Edge Flag Search
;
EFS:
	#(JMP+DONE)
;
; EFF - Edge Flag Fill
;
EFF:
	#(JMP+DONE)
;
; IFS - Interior Fill Search
;
IFS:
	#(JMP+DONE)
;
; IFF - Interior Fill Fill
;
IFF:
	#(JMP+DONE)
;
; OFS - Overlay Fill Search
;
OFS:
	#(JMP+DONE)
;
; OFF - Overlay Fill Fill
;
OFF:
	#(JMP+DONE)
;
; GFS - General Fill Search
;
GFS:
	#(JMP+DONE)
;
; GFF - General Fill Fill
;
GFF:
	#(JMP+DONE)

	.END	ZERO		; fini...

;
; Permanent symbol table - W08
;
; Consists of entries for each of the fields
; Defined as FLD NAME,VAL
;
	.MACRO	FLD,FIELD,CODE
	.NCHR	LEN,FIELD
	.ASCII	<LEN>/FIELD/<CODE>
	.ENDM
;
; Message printed at startup
;
HIMSG::	.ASCII	<15><12>'AED W08 Assembler '<200>
	.EVEN

; CONDITIONAL "IF" TERMS
IFTRM:	FLD	EQ,0
	FLD	NE,1
	FLD	DF,2
	FLD	NDF,3
	FLD	NEQ,4
	.BYTE	0
	.EVEN

;  JUMP CODES
;  BIT 4 = ADDR, BIT 3-0 = CODE
JSRC:	FLD	CJV,26
	FLD	CJS,21
	FLD	JMP,22
	FLD	CRTN,12
	FLD	JMAP,22
	FLD	CJP,23
	FLD	PUSH,24
	FLD	JSRP,25
	FLD	JRP,27
	FLD	RFCT,10
	FLD	RPCT,31
	FLD	CJPP,33
	FLD	LDCT,34
	FLD	LOOP,15
	FLD	CONT,16
	FLD	TEST,16
	FLD	TWB,37
	FLD	JZ,0
	.BYTE	0
	.EVEN
�
;  DESTINATION CODES
OPSRC:	FLD	Q,0
	FLD	F,1
	FLD	BUS,1
	FLD	NOP,1
	FLD	OUT,1
	FLD	STCK,1
	.BYTE	0
	.EVEN
;  MODIFIER CODES
MSRC:	FLD	Q,0
	FLD	NOP,1
	FLD	BUS,2
	FLD	OUT,2
	FLD	ROTR,4
	FLD	SR,5
	FLD	ROTL,6
	FLD	SL,7
	FLD	RAMQD,4
	FLD	RAMD,5
	FLD	RAMQU,6
	FLD	RAMU,7
	.BYTE	0
	.EVEN

;  1ST FUNCTION FIELD CODES
F1SRC:	FLD	Q,2
	FLD	Z,5
	FLD	D,6
	FLD	NOP,7
	FLD	INC,0+10
	FLD	DEC,1+10
	FLD	COM,7+10
	.BYTE	0
	.EVEN
;  2ND FIELD FUNCTION CODES
F2SRC:	FLD	BIS,3
	FLD	MASK,4
	FLD	BIC,5
	FLD	XOR,6
	FLD	XNOR,7
	FLD	Q,2+10
	FLD	D,6+10
	FLD	ADD,0
	FLD	SUBR,1
	FLD	SUBS,2
	FLD	OR,3
	FLD	AND,4
	FLD	NOTAND,5
	FLD	NOTN,5
	.BYTE	0
	.EVEN

;  3RD FUNCTION FIELD CODES
F3SRC:	FLD	Q,2
	FLD	Z,5
	FLD	D,6
	.BYTE	0
	.EVEN

;  CARRY MODIFIER CODES
FCSRC:	FLD	C,0
	FLD	CN,0
	.BYTE	0
	.EVEN
;
;	Default field positions
;

FLDVL:
.BCN:	.BYTE	48.
.FUN:	.BYTE	45.
.SRC:	.BYTE	42.
.DES:	.BYTE	40.
.AAD:	.BYTE	36.
.BAD:	.BYTE	32.
.BRC:	.BYTE	29.
.IDF:	.BYTE	16.
.SQC:	.BYTE	12.
.CCT:	.BYTE	10.
.SHF:	.BYTE	9.
.CLC:	.BYTE	3
.SEC:	.BYTE	2
.ICF:	.BYTE	4
.IMC:	.BYTE	0
.SEQ:	.BYTE	16
	.EVEN
	.END

			Introduction

This document presents the internal archetecture for the new firmware
for WINGS using the WDC65816 processor board. It details the instruction
set for the previous graphics terminals (512,767,1024) in sufficent
detail to enable future products to use this document as the standard
for "1024 compatibility" and also presents the internal design overview
of the WINGS software. This document will be organized such that sections
will be usable in future software support and user-level programmers
reference manuals.

